VIII olimpiada

I etapas

Jaunesniems (iki IX kl.)

85. Ženklo kaita. Ženklo kaitą galima tikrinti dviejų gretimų sekos narių sandauga: jei ji neigiama, vadinasi, šie nariai yra priešingų ženklų. Tačiau dviejų sveikųjų skaičių sandauga gali viršyti maxint.

Kiekvienam sekos nariui parinkime jam atitinkantį daugiklį. Kai narys teigiamas, daugiklis bus lygus 1, kai neigiamas lygus -1. Jei du gretimus sekos narius atitinkančių daugiklių sandauga neigiama, vadinasi, šioje vietoje seka keičia ženklą.



program ženklo_kaita;

  var keičia,           { kiek kartų sekoje keičiamas ženklas }

      a1, a2: integer;  { du gretimi sekos nariai }

      d1, d2: -1..1;    { juos atitinkantys daugikliai }

begin

  writeln ('Įveskite sveikųjų nenulinių skaičių seką:');

  keičia := 0;

  read (a1, a2);  { kadangi seka netuščia, tai a1 < > 0 }

  if a1 < 0

     then d1 := -1

     else d1 := 1;

  while a2 <> 0 do   { kol neperskaityta visa seka }

    begin

      if a2 < 0

         then d2 := -1

         else d2 := 1;

      if d1 * d2 < 0    { gretimi sekos nariai yra priešingų ženklų }

        then keičia := keičia + 1;

      a1 := a2;

      d1 := d2;

      read (a2);

   end;

  writeln (keičia)

end.



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�1 0�0�Seka nė karto nekeičia ženklo;��2�1 2 -3 4 4 -5 -1 9 -12 34 -9 -9 -3 87 34 -23  -19 0�9�Seka keičia ženklą kelis kartus;��3�-1 1 34 34 -6 4 -7 9 -3 33 33 -10 3 45 67 -9 12 9 100       -10000 -1 11 1 1 -1 1 -1 -1 2 -2 0�18�Atvejis, kai gretimų skaičių sandauga gali viršyti maxint.��

86. Keturženklių skaičių suma. Sprendžiant šį uždavinį reikėtų atkreipti dėmesį į skaitmenis:

1)  keli jų gali būti vienodi;

2)  tarp jų gali būti nulių.

Abiem atvejais sumažėja ieškomų skirtingų keturženklių skaičių.

Vienas sprendimo būdų būtų toks. Iš duotų skaitmenų sudaromas mažiausias galimas keturženklis skaičius ir perstatinėjant skaitmenis konstruojami visi galimi keturženkliai skaičiai didėjimo tvarka. Tai būtų pats efektyviausias sprendimo būdas. 

Tačiau keturženklių skaičių nėra daug. Jų perrinkimas neužims daug laiko.

Todėl pasirinksime paprastesnį sprendimą. Perrinksime visus keturženklius skaičius nuo 1000 iki 9999. Tuomet tikrai aišku, kad nė vienas skaičius nebus pridėtas du kartus. 

Patogiausia pradinius duomenis (keturis skaitmenis) surikiuoti didėjimo tvarka. Nagrinėjamąjį keturženklį skaičių taip pat suskaidyti į skaitmenis ir tuos skaitmenis surikiuoti didėjimo tvarka. Palyginę atitinkamus skaitmenis, sužinome, ar keturženklis skaičius sudarytas iš duotų skaitmenų, ar ne.

Ieškoma suma gali viršyti maxint, todėl funkcijos rezultatas yra longint tipo.

type skaitmuo = 0..9;



function keturženkliai(a, b, c, d: skaitmuo): longint;

  var sk1, sk2, sk3, sk4: skaitmuo;{ keturženklio skaitmenys }

      i: integer;

      suma: longint;



  procedure sukeisti (var x, y: skaitmuo);

  { sukeičia kintamųjų x ir y reikšmes vietomis }

    var tarp: skaitmuo;

  begin

    tarp := x;

    x := y;

    y := tarp

  end; { sukeisti }



  procedure rikiavimas (var a, b, c, d: skaitmuo);

  { surikiuoja keturis skaitmenis didėjimo tvarka }

  begin

    if a > b then sukeisti (a, b);

    if a > c then sukeisti (a, c);

    if a > d then sukeisti (a, d);

    if b > c then sukeisti (b, c);

    if b > d then sukeisti (b, d);

    if c > d then sukeisti (c, d)

  end; { rikiavimas }



begin  { keturženkliai }

  rikiavimas (a, b, c, d);

  suma := 0;

  for i := 1000 to 9999 do

    begin

      sk1 := i mod 10;       { išskaidome skaitmenimis }

      sk2 := i div 10 mod 10;

      sk3 := i div 100 mod 10;

      sk4 := i div 1000;

      rikiavimas (sk1, sk2, sk3, sk4);

      if (a = sk1) and (b = sk2) and 

         (c = sk3) and (d = sk4)

         then suma := suma + i

    end

  keturženkliai := suma

end; { keturženkliai }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�0 0 0 0�0�Visi skaitmenys nuliai;��2�0 0 9 0�9000�Trys skaitmenys nuliai;��3�0 5 6 0�34221�Du skaitmenys nuliai;��4�0 8 0 8�24888�Du skaitmenys lygūs nuliui, kiti du sutampa;��5�1 0 2 2�16110�Vienas skaitmuo lygus nuliui, du skaitmenys sutampa;��6�7 0 3 4�90216�Vienas skaitmuo nulis;��7�5 5 5 5�5555�Visi skaitmenys vienodi;��8�3 4 8 3�59994�Du skaitmenys vienodi;��9�7 4 6 5�146652�Visi skaitmenys skirtingi;��

87. Degtukai. Norint sukonstruoti iš vienodo ilgio degtukų stačiakampį, jų skaičius turi būti lyginis, t. y. dalus iš 2. Be to, reikėtų nepamiršti, kad turi būti ne mažiau kaip 4 degtukai. 

Kadangi uždavinio sąlygoje nereikalaujama nustatyti, kokią figūrą įmanoma sudėti, tai kvadrato galima nenagrinėti – tai atskiras stačiakampio atvejis. 

Lygiakraštį trikampį sudėlioti iš degtukų galima tik tada, kai degtukų skaičius dalijasi iš 3 (turi būti trys lygios kraštinės). 

Vadinasi, jei degtukų skaičius dalus iš 3, arba dalus iš 2 ir ne mažesnis kaip 4, tai galima sudėlioti bent vieną pageidaujamą figūrą. 

function galima (n: integer): boolean;

begin

  galima := (n >= 4) and (n mod 2 = 0) or

            (n mod 3 = 0)

end; { galima }



Testo nr.�Pradinis duomuo�Rezultatas�Paaiškinimai��1�1�Negalima�Per mažai degtukų;��2�2�Negalima�Per mažai degtukų;��3�3�Galima�Paprastas atvejis, kai galima sudėti lygiakraštį trikampį;��4�12�Galima�Galima sudėti visas figūras;��5�15�Galima�Galima sudėti tik trikampį;��6�16�Galima�Galima sudėti kvadratą ir stačiakampį;��7�35�Negalima�Negalima sudėti nei vienos figūros;��

Vyresniems (X – XII kl.)

88. Įdomi seka. Iš eilės generuojame natūraliųjų skaičių kvadratus. Kiekvieną kartą nustatome kvadrato skaitmenų skaičių ir jį atimame iš pradinio duomens. Šitaip "šliaužiame" į priekį iki tol, kol duomuo, iš kurio atimame, tampa mažesnis arba lygus nagrinėjamo kvadrato skaitmenų skaičiui. Tada nagrinėjamo skaičiaus kvadrate suieškome reikalingą skaitmenį.

Pradinis duomuo k negali būti labai didelis. Priešingu atveju jo kvadratas viršys maxint. Tada gausime sunkesnį, nebe pirmojo etapo lygio uždavinį, mat tektų modeliuoti veiksmus su dideliais skaičiais.

Didžiausias skaičius, kurio kvadratas neviršija maxint, yra 181. Tai ir bus didžiausia leistina k reikšmė.

type skaitmuo = 0..9;



function seka (k: integer): skaitmuo;

  var sk,       { natūraliųjų skaičių sekos nariai }

      kv,       { jų kvadratai }

      skaičius, { kvadrato skaitmenų skaičius }

      i, m: integer;



  function skait_sk (n: integer): integer;

    { randa skaičiaus n skaitmenų skaičių }

    var s: integer;

  begin

    s := 0;

    repeat

      s := s + 1;

      n := n div 10;

    until n = 0;

    skait_sk := s

  end; { skait_sk }



begin

  sk := 1;  { pradedame nuo pirmojo natūraliojo skaičiaus }

  skaičius := 1;

  while skaičius < k do  { ieškome kvadrato, kuriame būtų }

    begin                { reikalingas skaitmuo }

      k := k - skaičius;

      sk := sk + 1;

      skaičius := skait_sk (sk * sk)

    end;

  { rastame kvadrate ieškosime reikalingo skaitmens }

  kv := sk * sk;

  m := skaičius - k + 1; { tai bus m-asis nuo skaičiaus galo skaitmuo }

  for i := 1 to m-1 do

     kv := kv div 10;

  seka := kv mod 10

end; { seka }



Testo nr.�Pradinis duomuo�Rezultatas�Paaiškinimai��1�1�1�Ribinis atvejis;��2�5�6�Paprastas atvejis, rezultatą dar galima rasti ir be kompiuterio;��3�29�9�Du rimtesni atsitiktinai parinkti testai;��4�100�9���5�181�6�Didžiausias k, kai k2 neviršija maxint;��

89. Būdvardžių laipsniavimas. Šiam uždaviniui pirmiausia reikia žinoti lietuvių kalbos būdvardžių laipsniavimo taisykles.

Aukštesnysis ir aukščiausiasis laipsniai daromi iš nelyginamosios laipsnio formos su tam tikromis priesagomis. Vyriškosios ir moteriškosios giminės aukštesniojo laipsnio formos daromas su priesaga - esnis, -esnė (geresnis, geresnė, gražesnis, gražesnė), aukščiausiojo laipsnio formos – su priesaga -iausias, -iausia (geriausias, geriausia, gražiausias, gražiausia).

Daugelyje lietuvių kalbos gramatikų minimos dar vadinamosios visų aukščiausiojo laipsnio formos, kurias sudaro aukščiausiojo laipsnio forma su prisijungusiu prie jos įvardžiu pats, pati arba įvardžio daugiskaitos kilmininko forma visų. Pavyzdžiui, visų geriausias. Nutarkime, kad to mes neieškosime. Taip pat neieškosime ir aukštesniojo laipsnio mažybinių formų su priesagomis -ėlesnis, -ėlesnė.

Be to, nutarkime, kad duotas nelyginamojo laipsnio būdvardis yra neįvardžiuotinis (žinoma, programą galima nesunkiai modifikuoti, kad ji laipsniuotų ir įvardžiuotines būdvardžio formas). Taip pat neimsime ir bevardės giminės būdvardžių, nes būtų keblumų atpažinti juos nuo moteriškos giminės.

Taigi sutariame, kad sudarysime algoritmą laipsniuoti moteriškosios ir vyriškosios giminės neįvardžiuotiniams būdvardžiams, t. y. rasime jų aukštesnįjį ir aukščiausiąjį laipsnius.

Reikia numatyti dar keletą išimčių: tai būtina tuomet, kai nelyginamojo laipsnio būdvardžio kamienas baigiasi t, č arba d. Tuomet aukščiausiame laipsnyje vietoj t atsiranda č (aukštas – aukščiausias), o vietoj d – dž (saldus – saldžiausias). Aukštesniajame laipsnyje vietoj č atsiranda t (kurčia – kurtesnė).

Visa šita apžvelgus, belieka užrašyti patį algoritmą.

type būdvardis = string;



procedure laipsniavimas (nelyg: būdvardis;      { nelyginis }

                         var aukšt: būdvardis;  { aukštesnysis }

                         var aukšč: būdvardis   { aukščiausiasis });

  var k: integer; { žodžio ilgis }

      vyr: boolean;

      gal: integer;

begin

  k := length (nelyg);

  vyr := nelyg[k] = 's'; { nustatome giminę }

  { atmetame galūnę }

  if vyr    { vyriškoji giminė }

     then gal := 2

     else gal := 1;

  delete (nelyg, k-gal+1, gal);

  k := k - gal;

  { jei yra minkštumo ženklas – jį atmetame }

  if nelyg[k] = 'i'

     then begin

            delete (nelyg, k, 1);

            k := k - 1

          end;

  { sudarome aukštesnįjį laipsnį }

  if nelyg[k] = 'č'

     then aukšt := copy (nelyg, 1, k-1) + 't'

     else aukšt := nelyg;

  if vyr

     then aukšt := aukšt + 'esnis'

     else aukšt := aukšt + 'esnė';

  { sudarome aukščiausiąjį laipsnį }

  case nelyg[k] of

    't': aukšč := copy (nelyg, 1, k-1) + 'č';

    'd': aukšč := nelyg + 'ž';

    else aukšč := nelyg

  end;

  if vyr

     then aukšč := aukšč + 'iausias'

     else aukšč := aukšč + 'iausia';

end; { laipsniavimas }



Testo nr.�Pradinis duomuo�Rezultatai�Paaiškinimai��1�geras�geresnis, geriausias�Duotas sąlygoje��2�dailus�dailesnis, dailiausias�Duotas sąlygoje��3�žalias�žalesnis, žaliausias�Su minkštumo ženklu��4�daili�dailesnė, dailiausia�Moteriška giminė��5�kurčias�kurtesnis, kurčiausias�Keičiama č į t, su minkštumo ženklu��6�grublėtas�grublėtesnis, grublėčiausias�Triskiemenis, keičiama t į č��7�tvirtas�tvirtesnis, tvirčiausias�Keičiama č į t��8�gardus�gardesnis, gardžiausias�Keičiama d į dž��9�gardi�gardesnė, gardžiausia�Keičiama d į dž, moteriška giminė��

90. Kuoliukai. Šiame uždavinyje galima numatyti tris išskirtinius momentus, kuriuos būtina apgalvoti rašant algoritmą:

1) išgaubtos ir įgaubtos bangos „persikloja“;

2) gali iš viso nebūti išgaubtos ar įgaubtos bangos;

3) bangos galuose negali būti vienodo ilgio kuoliukų (žr. bangos apibrėžtį sąlygoje).

Kuoliukus numeruosime nuo 1 iki n. Panagrinėkime, kaip reikėtų ieškoti išgaubtos bangos. Paeiliui skaitome kuoliukų ilgius, kol randame du iš eilės einančius kuoliukus, kurių pirmasis būtų trumpesnis už antrąjį, t. y. hi < hi+1. Tai bus išgaubtos bangos pradžia. Įsimename šio kuoliuko numerį i (90.2.a pav.).

90.2.a pav.

Toliau skaitome kuoliukų ilgius. Bangos viršūnė bus pasiekta, kai surasime du iš eilės einančius kuoliukus, kurių pirmasis ilgesnis už antrąjį, t. y. hj > hj+1.

90.2.b pav.

Jei šitokių kuoliukų nepavyktų rasti iki tvorelės pabaigos, tai tiriamoji banga nebūtų išgaubta. 

90.2.c pav.

Pasiekę bangos viršūnę toliau skaitome kuoliukų ilgius. Kai rasime du gretimus kuoliukus, kurių pirmasis bus trumpesnis už antrąjį, t. y. hl < hl+1, tai reikš, kad prasidėjo nauja banga ir tiriamoji banga jau pasibaigė. Tačiau jei bangos pabaigoje yra vienodo ilgio kuoliukų, tai pagal sąlygą tik pirmasis jų priklauso tiriamajai bangai. Todėl šalia paskutinio kuoliuko numerio l reikia žinoti ir numerį kuoliuko ties kuriuo prasideda vienodų kuoliukų eilė. Jei baigiantis bangai vienodų kuoliukų nebuvo, tai šis numeris bus lygus l, o jei bus vienodų kuoliukų grupė, tuomet jis rodys į pirmąjį kuoliuką iš vienodų grupės.

90.2.d pav. 

Aptikus naujos bangos pradžią beliks įvertinti ištirtos bangos ilgį: mat pradžia buvo įsiminta anksčiau.

Dabar panagrinėkime atvejį, kai banga užsibaigia tvorelės pabaiga. Jei tvora baigiasi keliais vienodo ilgio kuoliukais, tuomet bangos pabaiga bus laikomas pirmasis iš vienodų kuoliukas. Taip būtų sąlygoje pateiktame pavyzdyje. 

Kitaip yra jei tvora baigiasi skirtingo aukščio kuoliukais (90.2 e pav.), tuomet pagal mūsų algoritmą bangos pabaiga būtų laikomas (j+1)-asis kuoliukas. Norėdami išvengti tokios situacijos, tvorelės gale prijungiame fiktyvų kuoliuką tokio pat ilgio kaip ir paskutinis.

90.2.e pav.

Tyrinėjant įgaubtą bangą skirtumas bus tik tas, kad vietoj gretimų kuoliukų ilgių padidėjimo bus sumažėjimas, o vietoj sumažėjimo – padidėjimas.

procedure bangos (var f: text;

                  n: integer;     { n >= 2 }

                  var  max_išg,  max_įg:  integer);

  var ilg1, ilg2,                 { gretimų kuoliukų ilgiai }

      i,                          { kuoliuko numeris [0..n-1] }

      pra_išg, pra_įg,            { išgaubtos ir įgaubtos bangų pradžios }


      vienodu_pr:      integer;
  
 { vienodo ilgio kuoliukų pradžia }


      vir_isg, vir_ig: boolean; 
  
{ pasiekta  bangos viršūn‚ }

Testo nr.�Pradiniai duomenys�Rezultatai�Paaiškinimai��1�5

6 6 6 6 6�0 0�Visi kuoliukai vienodo ilgio, nėra nei vienos bangos��2�15

15 15 15 14 13 12 11 10 9 8 7 6 5 5 5�0 0�Nėra nei vienos bangos��3�14

1 2 3 4 5 6 7 8 7 6 5 4 2 1�14 0�Yra vienintelė išgaubta banga��4�22

10 10 10 11 11 12 13 14 15 16 16 16 15 15 14 12 12 10 10 9 9 9 �18 0�Yra vienintelė išgaubta banga, joje yra vienodo ilgio iš eilės einančių kuoliukų��5�26

4 2 2 2 3 4 5 5 5 6 6 4 3 3 2 1 1 1 2 3 5 5 6 6 5 5�13 13�Šis testas atitinka sąlygoje pateiktą paveikslėlį��6�32767

1 2 3 4 ...�10 11�Kuoliukų skaičius labai didelis��

Pastaba. Kai kuriuose testuose pradiniai duomenys surašyti ne į dvi, o į daugiau eilučių.



II etapas

Jaunesniems (iki IX kl.)

91. Nutrinti skaičiai. Perrinkti visus iki vieno variantus būtų labai neefektyvu. Geriausia nagrinėti atskirus atvejus. Juos visus peržiūrėti nėra sunku. Iš viso yra penki variantai:

1) nutrinti skaičiai a ir s ; Tuomet a ( d div b, s ( a + b;

2) nutrinti skaičiai a ir d ; Tuomet a ( s - b, d ( a ( b;

3) nutrinti skaičiai b ir s ; Tuomet  b ( d div a, s ( a + b;

4) nutrinti skaičiai b ir d ; Tuomet b ( s - a, d ( a ( b;

5) nutrinti skaičiai s ir d ;   Tuomet s ( a + b, d ( a ( b;

Taigi nutrintieji skaičiai atstatomi keitimo būdu sprendžiant lygčių sistemą:

�               s = a + b,

               d = a ( b.

Labai svarbu pastebėti, kad jei skaičius s yra nutrintas, ši lygčių sistema ne visada turės sprendinius natūraliųjų skaičių aibėje. Galimi tokie du atvejai:

1) d mod b ( 0  (nežinomi a ir s);

2) d mod a ( 0  (nežinomi b ir s);

Natūraliojo sprendinio taip pat gali nebūti, kai, pavyzdžiui s > a.

Tačiau sąlygoje kalbama apie nutrintus skaičius. Taigi logiška manyti, kad sprendinys būtinai egzistuoja.

procedure nutrinti (var a, b, s, d: integer);

begin

  if s = 0

  then if d = 0

       then begin    { nutrinti skaičiai s ir d }

              s := a + b;

              d := a * b;

            end

       else begin { nutrinti skaičiai s ir   a arba b }

              if a = 0

                 then a := d div b

                 else b := d div a;

              s := a + b

            end

  else begin  { nutrinti skaičiai d ir   a arba b }

         if a = 0

            then a := s - b

            else b := s - a;

         d := a * b

       end

end; { nutrinti }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�0 12 0 48�4 12 16 48�Nutrinti skaičiai a ir s ; ��2�0 5 9 0�4 5 9 20�Nutrinti skaičiai a ir d ;��3�3 0 0 39�3 13 16 39�Nutrinti skaičiai b ir s ; ��4�15 0 105 0�15 90 105 1350�Nutrinti skaičiai b ir d ;��5�25 13 0 0�25 13 38 325�Nutrinti skaičiai s ir d ;��6�1 0 0 32766�1 32766 32766 32766�Rezultatai – skaičiai, artimi maxint;��

92. Savotiški skaičiai. Reikia peržiūrėti kiekvieną tiriamojo intervalo skaičių ir nustatyti, ar jis tenkina reikalaujamas sąlygas:

1) ar šis skaičius nėra palindromas;

2) ar šis skaičius dalijasi iš savo skaitmenų;

3) ar atvirkščia tvarka parašytas skaičius dalijasi iš tų pačių skaitmenų.

Ypač reikia atkreipti dėmesį į tai, kad dalyba iš nulio negalima, todėl būtų geriau pirmiausia patikrinti kiekvieną tiriamąjį skaičių, ar jame nėra lygių nuliui skaitmenų. Jei taip, tai šiam skaičiui nebeverta skirti dėmesio, – jis netinkamas.



program savotiški_skaičiai;



  function atbulas (sk: longint): longint;

  { skaičiaus sk perrašymas iš kito galo (atbulai) }

      var atb: longint;

  begin

    atb := 0;

    repeat

      atb := atb * 10 + sk mod 10;

      sk := sk div 10

   until sk = 0;

   atbulas := atb

  end; { atbulas }



  function nuliai (sk: longint): boolean;

  { ar tarp skaičiaus sk skaitmenų yra nulių }

    var yra: boolean;

  begin

    yra := false;

    while (sk <> 0) and not yra do

      begin

        yra := sk mod 10 = 0;

        sk := sk div 10

      end;

    nuliai := yra;

  end; { nuliai }



  function palindromas (sk: longint): boolean;

  { nustato, ar skaičius yra palindromas }

  begin

    palindromas := sk = atbulas(sk);

  end; { palindromas }



  function dalijasi (sk: longint): boolean;

  { nustato, ar skaičius sk dalijasi iš savo skaitmenų }

  { tarp skaitmenų tikrai nėra nulių }

    var gerai: boolean;

        prad: longint; { skaičiaus sk kopija }

  begin

    gerai := true;

    prad := sk;

    while (sk <> 0) and gerai do

      begin

        gerai := prad mod (sk mod 10) = 0;

        sk := sk div 10;

      end;

    dalijasi := gerai

  end; { dalijasi }



  var min, max, { n-ženklių skaičių intervalas }

      kiek,  { kiek yra skaičių }

      i, n, sk: longint;

begin

  readln(n);

  min := 1;

   for i := 2 to n do

      min := min * 10 + 1;

   max := min * 9;

   kiek := 0;

   for sk := min to max do

     if not nuliai(sk) and not palindromas(sk)

        then if dalijasi(sk) and dalijasi(atbulas(sk))

                then begin

                       writeln(sk);

                       kiek := kiek + 1

                      end;

  if kiek = 0

     then writeln('Nėra');

end. { savotiški_skaičiai }



Testo nr.�Pradinis duomuo�Rezultatai�Paaiškinimai��1�2�Nėra�Tokių skaičių iš viso nėra��2�3�216

612���3�4�1113

1131

...�Iš viso yra 64 tokie skaičiai��4�5�11133

11313

...�Iš viso yra 282 tokie skaičiai��

93. Žiogas. Žinome, kad virvutė yra s sprindžių ilgio. Tarkime, kad joje kas sprindį bei abiejuose galuose užrišta po mazgą. Mazgai sunumeruoti nuo 0 iki s. Susidarysime lentelę, kurioje žymėsimės kiek mažiausiai žingsnių reikia padaryti, norint pasiekti atitinkamą mazgą. Jei mazgas nepasiekiamas, lentelės elemente įrašytas -1. 

Pavyzdžiui, pradiniai duomenys yra tokie: s ( 11, c ( 5, a ( 3, b ( 1. Sudarome lentelę iš 12 elementų:

Mazgo nr.�0�1�2�3�4�5�6�7�8�9�10�11��Per kiek žingsnių jis pasiekiamas�0�-1�-1�-1�-1�-1�-1�-1�-1�-1�-1�-1��Pradžioje žiogas stovi ant virvutės krašto, taigi nulinis mazgas yra pasiektas. Peržiūrime lentelę, randame visus mazgus, kuriuose galima pasiekti per nulį žingsnių (šiuo atveju bus vienas toks mazgas) ir nuo jo žiogas šoka a sprindžių pirmyn. Atgal šokti žiogas negali, nes nukris nuo virvutės.

Mazgo nr.�0�1�2�3�4�5�6�7�8�9�10�11��Per kiek žingsnių jis pasiekiamas�0�-1�-1�1�-1�-1�-1�-1�-1�-1�-1�-1��Dabar peržiūrime visus mazgus, randame tuos, kuriuos žiogas gali pasiekti padaręs tik vieną šuolį ir nuo jų žiogas šoka pirmyn ir atgal.

Mazgo nr.�0�1�2�3�4�5�6�7�8�9�10�11��Per kiek žingsnių jis pasiekiamas�0�-1�2�1�-1�-1�2�-1�-1�-1�-1�-1��Taip pildome lentelę, kol pasiekiame mazgą, nutolusį nuo pradžios per c sprindžių arba žiogas nebegalės pasiekti nei vieno mazgo, kuriame jo nebūta. Baigus darbą su šiais pradiniais duomenimis, lentelė turėtų atrodyti taip:

Mazgo nr.�0�1�2�3�4�5�6�7�8�9�10�11��Per kiek žingsnių jis pasiekiamas�0�3�2�1�-1�3�2�-1�-1�3�-1�-1��

const MAX = 1000; { maksimalus virvutės ilgis }

type lentelė = array [0..MAX] of integer;



function šokti (s, a, b, c: integer): integer;

  var l: lentelė; { žymėsime pasiektus mazgus }

      i, žingsniai: integer;

      galima: boolean; { ar galima pasiekti naujas vietas virvutėje }

begin

  { nei vienoje virvutės vietoje žiogas dar nebuvo }

  for i := 1 to s do

    l[i] := -1;

  l[0] := 0;  { žiogas stovi ant virvutės galo }

  žingsniai := 0; { dar nepadarė nei vieno šuolio }

  galima := true;

  while galima and (l[c] = -1) do

    begin

      galima := false;

      { šoks į priekį }

      for i := 0 to s-a do

        if (l[i] = žingsniai) and (l[i+a] = -1)

            then begin

                   galima := true;

                   l[i+a] := žingsniai + 1

                 end;

      { šoks atgal }

      for i := b to s do

        if (l[i] = žingsniai) and (l[i-b] = -1)

           then begin

                  galima := true;

                  l[i-b] := žingsniai + 1

                end;

      žingsniai := žingsniai + 1

    end;

  šokti := l[c];

end; { šokti }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�10 3 2 6�2�Žiogas turi šokti du kartus į priekį��2�20 3 1 5�3�Žiogas šoks du kartus į priekį ir kartą atgal��3�20 4 2 5�Negalima�Mazgo pasiekti negalima��4�9 7 4 8�Negalima�Mazgą galima būtų pasiekti tik tada, jei virvutė būtų begalinio ilgio (tada reiktų 9 šuolių)��5�100 13 1 27�15�Žiogas šoka dideliais žingsniais į priekį ir mažais atgal��

Vyresniems (X – XII kl.)

94. Plokštelės su skylutėmis. Norint išspręsti šį uždavinį reikia mokėti dvimačius masyvus. Uždavinys nereikalauja kokio nors įmantresnio algoritmo. Tereikia kruopščiai atlikti patikrinimą. 

Reikia patikrinti aštuonis variantus: keturi variantai gaunami vis pasukant antrąją plokštelę 90 laipsnių kampu, ir keturi variantai – apvertus antrąją plokštelę.

Tikrinti galima dviem būdais:

1) sukti (versti) antrąją plokštelę ir tada ją dėti ant pirmosios;

2) iš karto skaičiuoti, kiek kiekvienu iš aštuonių atvejų liks neuždengtų skylučių. Tam reikia peržiūrinėti visus pirmosios plokštelės kvadratėlius. Radus skylutę patikrinti, kuriais iš aštuonių variantų skylutė bus uždengta. Jei pirmosios plokštelės skylutės koordinatės yra (i, j), tai galimus aštuonis variantus atitiks šie antrosios plokštelės kvadratėliai:

1) nepasukta plokštelė: (i, j);

2) pasukta 90( kampu pagal laikrodžio rodyklę (j, n - i + 1);

3) pasukta 180( kampu pagal laikrodžio rodyklę (n - i + 1, n - j + 1); 

4) pasukta 270( kampu pagal laikrodžio rodyklę (n - j + 1, i);

5) apversta ir nepasukta plokštelė: (i, n - j + 1);

6) apversta ir pasukta 90( kampu pagal laikrodžio rodyklę (n - j + 1, n - i + 1);

7) apversta ir pasukta 180( kampu pagal laikrodžio rodyklę (n - i + 1, j);

8) apversta ir pasukta 270( kampu pagal laikrodžio rodyklę (j, i);

94.1.1-8 pav.

const MAX = 100; { maksimalus plokštelės dydis }

type plokštelė = array [1..MAX, 1..MAX] of boolean;

                 { true – kvadratėlis iškirptas }



function skylutės (n: integer; var pl1, pl2: plokštelė)

                                               : integer;

  var s: array [1..8] of integer;

      { kiek skylučių matosi uždėjus antrąją plokštelę ant pirmosios }

      { atitinkamai apvertus ar pasukus antrąją plokštelę }

      i, j, min: integer;

begin

  for i := 1 to 8 do

    s[i] := 0;

  for i := 1 to n do

    for j := 1 to n do

       if pl1[i, j] { jei skylutė pirmajame kvadratėlyje }

       then begin

             if pl2[i, j] then s[1] := s[1] + 1;

             if pl2[j, n-i+1] then s[2] := s[2] + 1;

             if pl2[n-i+1, n-j+1] then s[3] := s[3] + 1;

             if pl2[n-j+1, i] then s[4] := s[4] + 1;

             if pl2[i, n-j+1] then s[5] := s[5] + 1;

             if pl2[n-j+1, n-i+1] then s[6] := s[6] + 1;

             if pl2[n-i+1, j] then s[7] := s[7] + 1;

             if pl2[j, i] then s[8] := s[8] + 1;

            end;

  min := s[1];

  for i := 2 to 8 do

    if s[i] < min

       then min := s[i];

  skylutės := min;

end; { skylutės }



Testo nr.�Rezultatas�Paaiškinimai��1�3�Antrąją plokštelę reikia pasukti 180( kampu;��2�8�Antrąją plokštelę reikia pasukti 270( kampu;��3�0�Antrąją plokštelę reikia apversti;��4�5�Antrąją plokštelę reikia apversti ir pasukti 90( kampu;��Pastaba. Testai pateikti 94.2.1-4 pav.

94.2.1-4 pav.,



95. Stogas. Matematikų olimpiados dalyviams šis uždavinys būtų lengvai įveikiamas. Pakanka prisiminti, kad jei per keturis taškus galima nubrėžti plokštumą, tai tuos taškus jungiančio keturkampio (lygiagretainio) įstrižainės taip pat bus toje plokštumoje. Jų sankirtos taškas dalins įstrižaines pusiau (kadangi virbai stovi kvadrato viršūnėse). Todėl, jeigu pastatytume dar vieną virbą, kurio viršūnė būtų įstrižainių sankirtos taške, tai jo ilgis turėtų būti

(a + c)/2 = (b + d)/2.

Galima įrodyti, kad stogas uždedamas tada ir tik tada, kai galioja lygybė

a + c = b + d.

Tam, kad nustatytume, ar galima perstatyti virbus taip, kad būtų tenkinama uždavinio sąlyga, pakanka jų ilgius surikiuoti didėjančiai (arba mažėjančiai)

k ( l ( m ( n

ir patikrinti, ar k + n = l + m.

Jei sąlyga netenkinama, tai stogo ant virbų uždėti negalima.

type virbas = (a, b, c, d);

     virbai = array [a..d] of integer; { virbų ilgiai }



procedure perstatymas (var v: virbai; { perstatyti virbai }

                       var gerai, { virbų perstatyti nereikia }

                       galima: boolean

                       { ar galima perstatyti, jei žinome, ką reikia });

  var i, j, min: virbas;

      virb: integer;

begin

  { tikrinsime, ar virbus perstatyti reikia }

  gerai := v[a] + v[c] = v[b] + v[d];

  if not gerai

     then begin

            { surikiuosime virbus ilgių didėjimo tvarka }

            for i := a to c do

               begin

                 min := i;

                 for j := succ(i) to d do

                    if v[min] > v[j]

                       then min := j;

                 { keičiame vietomis }

                 virb := v[i];

                 v[i] := v[min];

                 v[min] := virb

               end;

            { ilgiausias virbas turi būti virbas c }

            virb := v[c];

            v[c] := v[d];

            v[d] := virb;

            galima := v[a] + v[c] = v[b] + v[d];

          end;

end; { perstatymas }



Testo nr.�Pradiniai duomenys�Rezultatai��1�125 15 25 115�GALIMA PERSTATYTI TAIP: 15 25 125 115 ��2�43 500 1111 654�GALIMA��3�1234 320 3087 4001�GALIMA��4�1005 2010 1005 2010�GALIMA PERSTATYTI TAIP: 1005 1005 2010 2010��5�10 15 88 61�PERSTATYTI NEGALIMA��6�85 13 31 25�PERSTATYTI NEGALIMA��

96. Sieninis kalendorius. Prieš pradedant spręsti uždavinį, reikia pastebėti, kad pro iškirptą langelį ne visuomet matysis skaičius. Gali matytis ir tuščias langelis. 

Kaip ir daugelyje uždavinių su datomis bei kalendoriais, vienas pirmųjų darbų būna sužinoti kuria savaitės diena prasideda metai.

Žinodami tai galėsime dirbti toliau. Per metus sausio 1-moji pasislenka arba viena (jei metai paprastieji), arba dviem (jei metai keliamieji) dienomis. Pavyzdžiui, 1979 sausio 1-moji buvo pirmadienis. Kadangi 1979 metai nėra keliamieji, tai 1980 m. sausio 1-moji yra antradienis. Šie metai jau yra keliamieji, todėl 1981 m. sausio 1-moji pasislenka per dvi dienas: yra ketvirtadienį.

Atskaitos pradžia galime pasirinkti šio šimtmečio paskutiniuosius metus. Žinome, kad kitų (1999) metų sausio 1-moji bus penktadienį. Eidami atgal ligi duotųjų metų, sužinome, kuria savaitės diena prasideda metai. 

type metai = 1583..1999;

     mėnuo = 1..12;

     diena = 0..31; { langelis gali būti tuščias, tad turi būti ir 0 }

     data = record

              mt: metai;

              mn: mėnuo;

              dn: diena

            end;



function kel (met: metai): boolean;

{ ar metai keliamieji }

begin

  kel := (met mod 400 = 0) or

         (met mod 100 <> 0) and (met mod 4 = 0)

end; { kel }



function pirma_diena (met: metai): diena;

{ randa kokia savaitės diena prasideda metai }

 var dien: diena;

     m, i: metai;

     posl: integer;

begin

  { 1999 metus pasirinkome atskaitos metais }

  dien := 5; m := 1999;

  for i := m-1 downto met do

    begin

      if kel (i) { randame poslinkį }

         then posl := 2

         else posl := 1;

      if dien - posl < 1

         then dien := dien + 7 - posl

         else dien := dien - posl;

    end;

  pirma_diena := dien

end; { pirma_diena }



Žinodami kuria savaitės diena prasideda metai, sužinome kuria savaitės diena prasideda pirmais (duotasis) mėnuo ir kuria savaitės diena prasideda antrasis mėnuo, (t. y. mėnuo, kurio diena matysis pro iškirptą skylutę). 

function dmėn (met: metai; mėn: mėnuo): diena;

{ kiek dienų turi met metų mėnuo mn }

begin

  case mėn of

    4, 6, 9, 11 : dmėn := 30;

    2: if kel(met) then dmėn := 29

                   else dmėn := 28;

    else dmėn := 31

  end

end; { dmėn }



function pir_mėn_dien (met: metai; mėn: mėnuo; 

                         die: diena): diena;

{ randa kuria savaitės diena prasideda met metų mėn mėnuo }

{ kai žinoma, kuria savaitės diena prasideda metai met }

  var dien, i: integer;

begin

  dien := die;

  for i := 1 to mėn-1 do

    dien := dien + dmėn (met, i);

  if dien mod 7 = 0

     then pirma_mėn_dien := 7

     else pirma_mėn_dien := dien mod 7;

end; { pir_mėn_dien }



Sakykime, kad pirmasis mėnuo prasideda penktadienį, o antrasis antradienį. Iš žemiau pateikto pavyzdžio, matosi, kad visos antrojo mėnesio dienos bus pasislinkę į priekį per �5-2 ( 3 dienas (96.2.a-b pav.).

96.2.a-b pav.

Taigi, vietoj 21-os dienos matysis 24-oji. 

function iškirptas_langelis (d: data; skyl: integer)

                                                 : diena;

{ randa skaičių, kuris matosi pro iškirptą langelį }

  var met_dien, mėn1_dien, mėn2_dien: diena;

      post: integer;

begin

  { rasime kokia savaitės diena prasideda duotieji metai }

  met_dien := pirma_diena (d.mt);

  { rasime kokia savaitės diena prasideda duotasis mėnuo }

  mėn1_dien := pir_mėn_dien (d.mt, d.mn, met_dien);

  { rasime kokia savaitės diena prasideda mėnuo, kurio diena }

  { matysis pro iškirptą langelį }

  mėn2_dien := pir_mėn_dien (d.mt, d.mn+skyl, met_dien);

  { rasime postūmį }

  post := mėn1_dien - mėn2_dien;

  if ((d.dn + post) <= 0) or 

     (d.dn + post > dmėn (d.mt, d.mn+skyl))

     then iškirptas_langelis := 0

     else iškirptas_langelis := d.dn + post

end; { iškirptas_langelis }



Testo nr.�Pradiniai duomenys�Rezultatas�Kurios savaitės kuri diena iškirpta�Paaiškinimai��1�1988 1 25 1�29�5 savaitė.,

Pirmadienis�Keliamųjų metų paskutinioji vasario diena��2�1999 1 25 1�Tuščia�5 savaitė

Pirmadienis�Nekeliamieji metai neturi 29-tos dienos��3�1998 11 30 1�Tuščia�6 savaitė

Pirmadienis�Turėtų matytis gruodžio 35-oji (mėnesio pabaiga)��4�1991 4 1 7�Tuščia�1 savaitė

Pirmadienis�Lapkritis prasideda pirmosios savaitės penktadienį, pro skylutę matosi pirmosios savaitės pirmadienis��5�1999 3 11 8�11�2 savaitė

Ketvirtadienis�Atsitiktinis testas, kai matosi skaičius��6�1974 1 6 10�3�1 savaitė

Sekmadienis�Atsitiktinis testas, kai matosi skaičius��

III etapas

I dalis 

Jaunesniems (iki IX kl.)

97. Ciklinis skaičius. Tarkime, kad turime skaičių n. Patikrinti, ar jis ciklinis – nesunku. Skaičių išskaidome skaitmenimis ir užrašome į masyvą. Tuomet einame ratu kaip nurodyta sąlygoje, pasižymėdami visus skaitmenis, kuriuos aplankėme, kol kurį nors skaitmenį aplankome antrą kartą. Skaičius ciklinis, jei aplankytų skaitmenų kiekis lygus skaičiaus skaitmenų kiekiui ir jei tas antrą kartą aplankytasis skaitmuo yra pirmas skaičiaus skaitmuo.

Na o jei duotasis skaičius nėra ciklinis, teks perrinkti didesnius už jį skaičius, kol rasime tinkamą.

const ILGIS = 8;

type skaičius = array [1..ILGIS] of 0..9;

     lentelė = array [1..ILGIS] of boolean;

                     { skaitmenų, įeinančių į seką lentelė }



function ciklinis (sk: skaičius; kiek: integer): boolean;

{ ar skaičius, turintis „kiek“ skaitmenų yra ciklinis }

  var būta: lentelė;

      nariai, { kiek skaitmenų įtraukta į seką }

      i: integer;

begin

  for i := 1 to kiek do

    būta[i] := false;  { seka kol kas tuščia }

  nariai := 0;         { ji neturi nei vieno nario }

  i := 1;

  while not būta [i] do

    begin

     būta[i] := true;

     nariai := nariai + 1;

     i := (i+sk[i]-1) mod kiek + 1;{ pereiname per sk[i] skaitmenų }

    end;

  ciklinis := (nariai = kiek) and (i = 1)

end; { ciklinis }



procedure užrašyti (n: longint;

                    var sk: skaičius; var kiek: integer);

{ skaičių n užrašo masyvu ir randa jo skaitmenų kiekį }

  var i, nn: longint;

begin

  { randame n skaitmenų skaičių }

  kiek := 0;

  nn := n;

  while nn > 0 do

    begin

      kiek := kiek + 1;

      nn := nn div 10

     end;

  { skaičių užrašome masyvu }

  for i := kiek downto 1 do

    begin

      sk[i] := n mod 10;

      n := n div 10

    end

end; { užrašyti }



function rasti (n: longint): longint;

{ randa mažiausią ciklinį skaičių, ne mažesnį už n }

  var sk: skaičius;

      kiek: integer; { kiek skaičius turi skaitmenų }

begin

  n := n - 1;

  repeat

    n := n + 1;

    užrašyti (n, sk, kiek);

  until ciklinis (sk, kiek);

  rasti := n;

end; { rasti }

98. Reiškinys. Kiekvieną reiškinį sudaro skaičiai, kintamieji bei tarp jų surašytos aritmetinės operacijos. 

Pradinis duomuo – simbolių eilutė. Tikriname jos pirmąjį simbolį – jei skaitmuo, vadinasi, reiškinys prasideda skaičiumi, jei raidė – kintamuoju. Tuomet nustatome objekto (skaičiaus ar kintamojo) ilgį, patį objektą pašaliname iš eilutės. 

Po skaičiaus ar kintamojo eina aritmetinė operacija. Ji sudaryta arba iš vieno simbolio (+, -, *, /), arba iš trijų (div, mod). Pašaliname iš simbolių eilutės operaciją ir toliau kartojame veiksmus – tikriname, ar likęs reiškinys prasideda skaičiumi, ar kintamuoju. 

Reikia nepamiršti keleto dalykų:

1)  reiškinio pradžioje esantis minusas rodo neigiamą skaičių, o ne aritmetinę operaciją;

2)  pradinis duomuo baigiasi lygybe; 

3)  po paskutiniojo objekto (kintamojo ar skaičiaus) nebebus aritmetinės operacijos;

4)  reiškinyje gali būti tarpų, todėl iš simbolių eilutės pašalinus objektą, reikia pašalinti ir po jo einančius tarpus;

procedure tarpai (var s: string);

{ pašalina tarpus iš reiškinio s pradžios }

begin

  while (s[1] = ' ') do

    delete (s, 1, 1)

end; { tarpai }



procedure skaičiuoti (s: string; { pradinis reiškinys }

                      var sk, kint, op: integer);

                     { kiek reiškinyje yra skaičių, kintamųjų ir operacijų }

{ suskaičiuoja kiek reiškinyje yra skaičių, kintamųjų ir operacijų }

  var ilgis: integer;

begin

  tarpai(s); { pašalinami tarpai reiškinio priekyje }

  { numetamas minusas reiškinio priekyje }

  if s[1] = '-'

     then delete (s, 1, 1);

  tarpai (s);

  sk := 0;

  kint := 0;

  op := 0;

  while s <> '=' do  { kol neišnagrinėjome viso reiškinio }

    begin

      { reiškinys prasidės arba skaičiumi, arba kintamojo vardu }

      if s[1] in ['0'..'9']

      then begin { randamas skaičiaus ilgis }

             skaičius (s, ilgis); 

             sk := sk + 1;

           end

      else begin { randamas kintamojo ilgis }

             kintamasis (s, ilgis); 

             kint := kint + 1;

           end;

      { pašaliname tą skaičių ar kintamąjį }

      delete (s, 1, ilgis);

      tarpai(s);

      { jei tai ne reiškinio pabaiga, randame aritmetinę operaciją }

      if s <> '='

      then begin

             if s[1] in ['+', '-', '*', '/']

                then ilgis := 1

                else ilgis := 3; 

             { operacijos div arba mod }

             delete (s, 1, ilgis);

             op := op + 1

           end;

      tarpai(s);

    end

end; { skaičiuoti }



Reiškinys gali prasidėti skaičiumi. Sveikąjį skaičių sudaro tik skaitmenys, realųjį – gali sudaryti skaitmenys ir simboliai „+“, „-“, bei raidė E (arba e):

0.15

0.89e-10

0.578E+3

7e-8

48.3e6



Skaičiaus pabaigos ieškome taip: kol nagrinėjami simboliai yra skaitmenys arba ‘.’, einame toliau. Jei aptikome kitokį simbolį ir tas simbolis nėra raidė E (e), radome skaičiaus pabaigą. Jei radome simbolį E (e), tuomet praleidžiame vieną simbolį (tai bus arba laipsnio rodiklio ženklas, arba paties rodiklio pirmasis skaitmuo) bei visus po to einančius skaitmenis.



procedure skaičius (s: string; var ilgis: integer);

{ s – reiškinys, prasidedantis skaičiumi }

{ procedūra randa jo ilgį }

begin

  ilgis := 2;

  while s[ilgis] in ['0'..'9', '.'] do

    ilgis := ilgis + 1;

  { gal būt tai dar ne skaičiaus pabaiga, o realaus skaičiaus laipsnis }

  if s[ilgis] in ['e', 'E']

     then begin

            ilgis := ilgis + 2;

            while s[ilgis] in ['0'..'9'] do

              ilgis := ilgis + 1

          end;

  ilgis := ilgis - 1

end; { skaičius }



Jei reiškinys pasideda kintamuoju, rasti reiškinio pabaigą nesunku – tereikia rasti pirmąjį simbolį, kuris nėra nei raidė, nei skaitmuo. Mat tarp kintamojo ir aritmetinių operacijų div ar mod būtinai turi būti paliktas tarpas (kitaip bus laikoma kintamojo vardu).

Reikia nepamiršti, kad kintamojo vardą gali sudaryti ne tik mažosios, bet ir didžiosios raidės.

procedure kintamasis (s: string; var ilgis: integer);

{ s – reiškinys, prasidedantis kintamuoju }

{ procedūra randa jo ilgį }

begin

  ilgis := 1;

  while s[ilgis] in ['A'..'Z', 'a'..'z', '0'..'9'] do

    ilgis := ilgis + 1;

  ilgis := ilgis - 1

end; { kintamasis }



Testo nr.�Pradiniai duomenys�Rezultatas��1�5+6-dc*a-0/b+15832-19+TARP2*169=�6 4 9��2�6 div 11 + abba mod e / 5 * 32740 =�4 2 5��3�8 + 5 div div6 mod moda=�2 2 3��4�-1682+16=�2 0 1��5�tarp=�0 1 0��6�a+8.00134-b*6.1348=�2 2 3��7�1e-5+0.8e+2-25.3e1=�3 0 2��8�1+1-1+1-1+1...1+1-1+1-1+1-1+1=�97 0 96��



Vyresniems (X – XII kl.)

99. Šaškės: viena prieš vieną. Panagrinėkime, kada baltoji gali nukirsti juodąją. 

Baltoji šaškė gali kirsti juodąją tik tada, kai abi šaškės stovi gretimose eilutėse, t. y. kai šaškes skiria 0 tuščių eilučių. 

1) Tarkime, kad abi šaškes skiria lyginis eilučių skaičius, pavyzdžiui 4. Atlikus baltajai šaškei vieną ėjimą, jas skirs 3 eilutės, atlikus juodajai šaškei ėjimą – dvi. T y. kiekvieną kartą, kai ateis baltosios šaškės eilė, jas skirs lyginis eilučių skaičius: 4, 2, 0. Juodoji šaškė niekada neturės galimybės numušti baltosios. Taigi, tokiu atveju baltoji šaškė pasieks pergalę arba lygiąsias (99.2 pav.).

99.2 pav.

2) Jei abi šaškes skiria nelyginis eilučių skaičius, baltoji šaškė niekaip negalės numušti juodosios. Šiuo atveju ji gali tik pasiekti lygiąsias arba pralaimėti.

Pagal šiuos samprotavimus ir sukursime strategiją: jei šaškes skiria lyginis eilučių skaičius, baltoji šaškė stengiasi artėti prie juodosios, jei nelyginis – nutolti.

Reikia nepamiršti poros atskirų atvejų. Jei baltoji šaškė stovi kraštiniame stulpelyje, ji turi tik vieną galimą ėjimą. Jei abi šaškės stovi toje pačioje eilutėje, ir jas skiria vienas langelis, baltoji turi bėgti nuo juodosios šaškės, nes toji gali kirsti atgal (99.3 pav.).

99.3 pav.

Turint šią strategiją pats sunkiausias dalykas, surasti sąryšį tarp langelio numerio ir eilutės bei stulpelio, kuriame jis yra. Algoritme laikėme, kad eilutės numeruojamos iš viršaus į apačią, stulpeliai – iš kairės į dešinę.



function ėjimas (b, j: integer): integer;

  var b_eil, b_st, j_eil, j_st: integer;

begin  { apskaičiuojame abiejų šaškių eilutes ir stulpelius }

  b_eil := (b - 1) div 5 + 1;

  if (b - 1) mod 10 < 5

     then b_st := b mod 5 * 2

     else b_st := 2 * (b mod 5 - 1) + 1;

  j_eil := (j - 1) div 5 + 1;

  if (j - 1) mod 10 < 5

     then j_st := j mod 5 * 2

     else j_st := 2 * (j mod 5 - 1) + 1;

  { rasime ėjimą }

  if b_st = 1  { jei baltoji šaškė kraštiniame stulpelyje }

       then b_st := b_st + 1

       else if b_st = 10

               then b_st := b_st - 1

  else if b_eil = j_eil  { abi šaškės vienoje eilutėje }

          then if b_st < j_st         { baltoji kairiau juodosios }

                  then b_st := b_st - 1    { tolstama }

                  else b_st := b_st + 1

  else if abs(b_eil-j_eil-1) mod 2 = 0 { lyginis eilučių skaičius }

          then if b_st < j_st

                  then b_st := b_st + 1     { artėjama }

                  else b_st := b_st -1

          else if b_st < j_st          { nelyginis eilučių skaičius }

                  then b_st := b_st - 1

                  else b_st := b_st + 1;  { artėjama }

  b_eil := b_eil - 1;

  if b_st mod 2 = 0

  then ėjimas := (b_eil-1) div 2 * 10 + b_st div 2

  else ėjimas := (b_eil-1)div 2 * 10 + 5 + b_st div 2 + 1

end; { ėjimas }

100. Kvadratinė šaknis. Jei reikėtų ištraukti šaknį iš nedidelio natūraliojo skaičiaus, turbūt naudotume dalijimo pusiau metodą. Tarkime, kad turime skaičių x. Akivaizdu, kad jo šaknis priklauso intervalui [1..x]. Padalijame šį intervalą į dvi dalis, imame intervalo vidurio tašką ab, ir keliame kvadratu. Jei gautasis skaičius didesnis už x, tai ieškoma šaknis yra pirmoje intervalo pusėje, jei mažesnis – antroje. Taip gavome dvigubai trumpesnį intervalą, kurį vėl dalijame į dvi dalis. Galiausiai intervale beliks du skaičiai. Arba vienas jų bus ieškoma šaknis, arba taps aišku, kad sveikos šaknies iš x ištraukti negalima. Šis metodas aprašytas [1], o panašiu algoritmu pasinaudota ir 60 uždavinyje.

Tačiau duotasis skaičius gali turėti iki 80 skaitmenų. Tai gerokai didesnis už maxlongint skaičius.  Didelių skaičių modeliavimas bei veiksmai su jais aprašyti [2]. Skaičių teks modeliuoti kitomis duomenų struktūromis, pavyzdžiui masyvu. Kiekvienas masyvo elementas yra skirtas vienam skaitmeniui. Šaknį galima traukti tik iš teigiamo skaičiaus. Jei duotasis skaičius galėtų būti neigiamas, tuomet skaičių tektų vaizduoti įrašu, kur vienas įrašo laukas būtų skaičiaus ženklas, kitas – skaičiaus modulis:

const SK_ILG = 80; { maksimalus skaičiaus ilgis }

type didelis = record

                  ženklas: (plius, minus);

                  modulis: array [1..SK_ILG] of 0..9;

                end;



Mūsų atveju užtenka paprasto masyvo. Maksimalų skaičiaus ilgį reikėtų padidinti. Mat keliant šaknies artinį kvadratu, galime gauti skaičių, turintį daugiau nei 80 skaitmenų.

  const SK_ILG = 90; { maksimalus skaičiaus ilgis }

  type didelis = array [1..SK_ILG] of 0..9;



Peržiūrėję kvadratinės šaknies traukimo algoritmą (60 uždavinys), pamatysime, kad su dideliais skaičiais reikės atlikti sudėties, atimties, daugybos, sveikųjų skaičių dalybos (div) bei lyginimo operacijas. 

Pradėkime nuo sudėties operacijos. Skaičiai sudedami stulpeliu.

procedure suma (n: integer; sk1, sk2: didelis; 

                var sum: didelis);

{ sudeda du teigiamus didelius skaičius, turinčius iki n skaitmenų }

  var perk, tarp, i: integer;

begin

  perk := 0;              { perkėlimas }

  for i := n downto 1 do  { pradedama nuo paskutiniojo skaitmens }

    begin

      tarp := sk1[i] + sk2[i] + perk;

      sum[i] := tarp mod 10;

      perk := tarp div 10;

    end;

end; { suma }



Atimties operaciją galime pakeisti sudėtimi. Vietoj to, kad tikrintume, ar b - a > 1, galime tikrinti, ar b > a + 1. 

Dauginsime stulpeliu, kaip ir įprasta matematikoje, pirmąjį skaičių dauginsime iš kiekvieno antrojo skaičiaus skaitmeniu ir gautas sandaugas sumuosime.

procedure sandauga (n: integer; sk1, sk2 : didelis; 

                    var san :didelis);

{ san := sk1 * sk2, kai daugikliai ir rezultatas turi iki n skaitmenų }

  var i, j, k,laik,  tarpinis : integer;

begin

  for i := n downto 1 do

    san[i] := 0;

  for i := n downto 1 do     { kiekvieną skaitmenį }

    for j := n downto 1 do   { dauginame iš antrojo skaičiaus }

      begin

        k := j + i - n;

        tarpinis := sk1[i] * sk2[j];

        while tarpinis <> 0 do  { tarpinę sandaugą pridedame prie }

          begin                 { turimos }

            laik := san[k];

            san[k] := (san[k] + tarpinis) mod 10;

            tarpinis := (laik + tarpinis) div 10;

            k := k - 1;

          end

      end;

end; { sandauga }



Dalinti sveikuosius skaičius reiktų kampu. Tačiau vienintelė algoritme atliekama dalybos operacija yra dalyba iš dviejų. Padalinti iš dviejų galima kur kas paprasčiau. Dalinti pradedama nuo galo. Rezultato skaitmuo bus lygus atitinkamam dalijamo skaičiaus skaitmeniui, padalintam iš dviejų. Jei skaitmuo nesidalija iš dviejų be liekanos, tuomet prie ankstesnio skaitmens pridedama 5. Padalinkime iš dviejų 3594102 (100.1 pav.):

100.1 pav.

Gavome 1797051

procedure div_2 (n: integer; sk: didelis; 

                 var dalmuo: didelis);

{ dalmuo := sk div 2; sk turi iki n skaitmenų }

  var i: integer;

begin

  dalmuo[n] := sk[n] div 2;

  for i := n-1 downto 1 do

    begin

      dalmuo[i] := sk[i] div 2;

      if sk[i] mod 2 = 1

         then dalmuo[i+1] := dalmuo[i+1] + 5;

    end;

end; { div_2 }



Algoritme vartojamos dvi lyginimo operacijos: daugiau ir daugiau arba lygu. Jas abi patogu pakeisti operacija mažiau. 

function mažiau (n: integer; sk1, sk2: didelis): boolean;

{ ar sk1 < sk2, kai skaičiai turi iki n skaitmenų }

  var i: integer;

begin

  i := 1;

  while (i < n) and (sk1[i] = sk2[i]) do

    i := i + 1;

  mažiau := sk1[i] < sk2[i];

end; { mažiau }



Sumodeliavę veiksmus su dideliais skaičiais, galime rašyti šaknies traukimo procedūrą. Traukiant šaknį iš nedidelio skaičiaus x, šaknies ieškoma intervale [1..x]. Šiuo atveju reikia stengtis intervalą kaip galima sumažinti. Jei skaičius turi n skaitmenų, tai jo šaknis turi (n - 1) div 2 + 1 skaitmenį, t. y. tiek, kiek didžiausias dešimties laipsnis, mažesnis už šį skaičių. Tai galite nesunkiai parodyti paeksperimentavę su dešimties laipsniais.

Kitas dalykas dėl ko reikia susitarti, tai – didelio skaičiaus laikymas masyve. Tarkime, kad dideliam skaičiui skiriama dešimt skilčių, o mes turime triženklį skaičių. Tuomet galimi du variantai. Arba skaičių dėti į paskutines skiltis, jo priekį užpildyti nuliais ir dirbti su visais 10 skaitmenų:

1�2�3�4�5�6�7�8�9�10��0�0�0�0�0�0�0�8�4�1��Arba skaičių saugoti į pirmosiose trijose skiltyse, likusias skiltis palikti neapibrėžtas ir dirbti tik su trimis skiltimis:

1�2�3�4�5�6�7�8�9�10��8�4�1���������

Dažniausiai yra sutinkamas pirmasis atvejis. Tačiau įdomumo dėlei savo algoritme pasirinkome antrąjį. Jei skaičius turi n skaitmenų, tai keliant šaknies artinį kvadratu, gali prireikti dirbti su skaičiais, turinčiais n+1 skaitmenį. Pavyzdžiui, traukiame šaknį iš 625. Žinome, kad šaknis yra dviženklis skaičius. Imame intervalo [10..99] vidurio tašką 54, keliame kvadratu ir gauname 2916 – keturženklį skaičių. Todėl skaitydami pradinius duomenis prieš skaičių įrašome nulį. Pavyzdžiui, skaičių 2916 įrašytume taip:

1�2�3�4�5�6�7�8�9�10��0�2�9�1�6�������

ir dirbtume su pirmomis n+1 skiltimis.

procedure šaknis (n: integer; { x skaitmenų skaičius }

                  x: didelis; { didelis skaičius įrašytas į n+1 skiltį }

                  var galima: boolean; { ar galima ištraukti šaknį }

                  var rez: didelis  { šaknis });

  var sk,   { skaitmenų skaičius ieškomoje šaknyje }

      i: integer;

      a, b, ab, { intervalo rėžiai ir vidurio taškas }

      a_plius_1, a_plius_b, vienas,

      xx: didelis;  { vidurio taškas, pakeltas kvadratu }

begin

  n := n + 1; { bus dirbama su n+1 skiltimi }

  for i := 1 to n-1 do

    vienas[i] := 0;

  vienas[n] := 1;               { a := 1 }

  sk := (n - 2) div 2 + 1;      { randame šaknies skaitmenų skaičių }

  for i := n downto 1 do        { sudarome intervalą }

    begin                       { a := 1000...000; }

      a[i] := 0;                { b := 9999...999; }

      if i >= n - sk + 1

         then b[i] := 9

         else b[i] := 0;

    end;

  a[n - sk + 1] := 1;

  sandauga (n, a, a, xx);

  if mažiau (n, xx, x) or mažiau (n, x, xx) { jei a nėra šaknis }

  then begin

         suma (n, a, vienas, a_plius_1); { sudarome a_plius_1 }

         while mažiau (n, a_plius_1, b) do { while b - a > 1 do }

           begin

             suma (n, a, b, a_plius_b);    { a_plius_b := a + b }

             div_2 (n, a_plius_b, ab);   { ab := a_plius_b div 2 }

             sandauga (n, ab, ab, xx);        { xx := ab * ab }

             if mažiau (n, xx, x)             { if xx < x }

                then a := ab                   { then a := xx }

                else b := ab;                  { else b := xx }

             suma (n, a, vienas, a_plius_1) { a_plius_1 := a+1}

           end

       end

  else b := a;  { jei a yra šaknis }

  sandauga (n, b, b, xx);                { xx := ab * ab }

  galima := not mažiau(n,xx,x) and not mažiau(n,x,xx);

  if galima

     then rez := b;

end; { šaknis }



Šiek tiek efektyvesnis šaknies traukimo algoritmas būtų vadinamasis šaknies traukimas kampu. Tačiau šį algoritmą žino stipriau besimokantys matematiką. Be to jame vis tiek reikia modeliuoti didelius skaičius. Kadangi galima apsieiti ir be labai matematiško sprendimo, jo nenagrinėjame.

Jaunesniems (iki IX kl.)

101. Stačiakampis. Tai tipiškas perrinkimo uždavinys. Vienintelė problema – optimizuoti perrinkimą, kad programos vykdymo laikas neviršytų duotojo.

Lotyniškoje abėcėlėje yra 26 raidės. Taigi, ieškomas stačiakampis negali turėti daugiau kaip 26 langelius. Naudinga suskaičiuoti, kiek yra skirtingų raidžių duotajame stačiakampyje: jų skaičiaus taip pat negali viršyti ieškomo stačiakampio dydis.

Perrenkama tokiu būdu. Sakykim, reikia surasti stačiakampį, kurio dydis yra 26. Peržiūrimi to skaičiaus dalikliai ir randamos galimos stačiakampio aukščio ir pločio kombinacijos. Šiuo atveju tai būtų (1, 26), (2, 13), (13, 2), (26, 1). Tuomet lape bandoma rasti kiekvieno dydžio stačiakampį.

Fiksuotų matavimų stačiakampio ieškoma taip. Imamas lapas ir laikoma, kad ieškomo stačiakampio kairiojo viršutiniojo kampo koordinatės yra (1, 1). Tuomet tikrinama, ar pažymėtame stačiakampyje visos raidės skirtingos. Jei ne, imama jo dalis su skirtingomis raidėmis ir žiūrima, ar ji nėra didesnė už iki šiol rastą didžiausią stačiakampį. Neradus ieškomo dydžio stačiakampio, viršutiniojo kampo koordinatės paslenkamos per vieną langelį į (1, 2) ir taip kartojami veiksmai, kol peržiūrimas visas lapas.

Pavyzdžiui, lapo dydis yra 4(5. Jame ieškome 2(3 dydžio stačiakampio �(101.1.a-d pav.). 

101.1.a-d pav.

Imkime atvejį d). Sakykime, kad jau peržiūrėjome pajuodintas raides ir sustojome ties antrąją F. Nors neradome stačiakampio, kurio dydis būtų lygus 6, radome du mažesnius stačiakampius. Vieno jų dydis lygus dviem, kito – trim. Jei iki šiol buvome radę mažesnį stačiakampį, įsimename didesnįjį iš rastųjų.

const MAX = 100; { maksimalus lapo kraštinės dydis }

      RSK = 26; { raidžių skaičius abėcėlėje }

type lapas = array [1..MAX, 1..MAX] of char;



{ žemiau surašyti globalieji kintamieji }

var s: lapas;      { duotasis lapas }

    m, n: integer; { jo išmatavimai }



function skirt: integer;

{ randa skirtingų raidžių skaičių stačiakampyje s }

  var aibė: set of char;

      i, j, kiek: integer;

      c: char;

begin

  { visas stačiakampio raides surašysime į aibę }

  aibė := [];

  for i := 1 to m do

    for j := 1 to n do

      aibė := aibė + [s[i, j]];

  { rasime skirtingų raidžių skaičiaus }

  kiek := 0;

  for c := 'A' to 'Z' do

    if c in aibė

       then kiek := kiek + 1;

  skirt := kiek

end; { skirt }



function daliklis (sk, x: integer): integer;

{ randa mažiausią sk daliklį didesnį už x }

begin

  x := x + 1;

  while  sk mod x <> 0 do

    x := x + 1;

  daliklis := x;

end; { daliklis }



procedure ieškoti (au, pl: integer;

                   var aukštis, plotis, 

                       eil, st, dydis: integer);

                    { jei randamas stačiakampis įsimenami jo duomenys }

{ tikrina, ar lape yra stačiakampis su skirtingomis raidėmis, }

{ kurio aukštis au, o plotis – pl }

  var i, j, i1, j1: integer;

      nerasta, skirtingos: boolean;

      aibė: set of char;

begin

  i := 1;  nerasta := true;

  while (i <= m-au+1) and nerasta do

    begin { kairysis viršutinis kampas yra eilutėje i }

      j := 1;

      while (j <= n-pl+1) and nerasta do

        begin   { ir stulpelyje j }

          { tikrinama, ar stačiakampis su tokiais parametrais }

          { sudarytas iš skirtingų raidžių }

          aibė := [];  skirtingos := true;

          i1 := i;

          while (i1 <= i+au-1) and skirtingos do

            begin

              j1 := j;

              while (j1 <= j+pl-1) and skirtingos do

                if s[i1, j1] in aibė

                   then skirtingos := false

                   else begin

                          aibė := aibė + [s[i1, j1]];

                          j1 := j1 + 1;

                        end;

               if skirtingos

                  then i1 := i1 + 1;

             end; { while }

          if skirtingos  { radome ko ieškojome }

             then begin  nerasta := false;

                         aukštis := au; plotis := pl;

                         eil := i; st := j; 

                         dydis := au * pl;

                  end

             else begin  { radome ne tokį didelį stačiakampį kokio }

                         { ieškojome, bet gal didesnį nei iki šiol rastą }

                    { rasime paskutinės skirtingos raidės koordinatę }

                    if j1 = 1

                       then begin i1 := i1 - 1;

                                  j1 := j+pl-1

                            end

                       else j1 := j1 - 1;

                    if (i1-i+1) * (j1-j+1) > dydis

                    then begin

                           aukštis := i1-i+1; 

                           plotis := j1-j+1;

                           eil := i; st := j;

                           dydis := aukštis * plotis;

                         end

                    else if (i1-i)*pl > dydis

                      then begin

                         aukštis := i1-i; plotis := pl;

                         eil := i; st := j;

                         dydis := aukštis * plotis;

                       end

                  end;

          j := j + 1;

        end;

      i := i + 1;

    end;

end; { ieškoti }



procedure rasti (var aukštis, plotis, eil, st: integer);

                 { ieškomo stačiakampio išmatavimai ir kairiojo }

                 { viršutinio kampo koordinatės }

{ randa didžiausią stačiakampį su skirtingomis raidėmis }

  var  didž, dydis, au, pl, i, j: integer;

begin

  { rasime skirtingų raidžių skaičių stačiakampyje }

  didž := skirt;       { didesnio stačiakampio negali būti }

  dydis := 1;          { didžiausio iki šiol rasto stačiakampio dydis }

  eil := 1;  st := 1;  { ir kiti duomenys }

  aukštis := 1;   plotis := 1;

  { atliksime perrinkimą }

  while dydis < didž do    { ieškome stačiakampio, kurio plotas – didž }

    begin

      pl := 1;

      while pl <= didž do  { tokio stačiakampio plotis yra pl }

        begin

          au := didž div pl;

          if (au <= m) and (pl <= n) then 

            ieškoti(au,pl,aukštis,plotis,eil,st,dydis);

          { randame tolesnę pl reikšmę }

          if pl < didž

             then pl := daliklis (didž, pl)

             else pl := pl + 1; { ciklas baigiasi }

        end;

      didž := didž - 1;

    end;

end; { rasti }

102. Kam patikėti paslaptį. Draugų pažintis galime pavaizduoti grafu (102.1 pav.).

102.1 pav.

Taigi norint sužinoti kiek paslapčių galima patikėti asmenų grupei tereikia sužinoti, kiek nejungių komponentų turi grafas.

Tai galima įvairiai sužinoti. Pasirinksime kurį nors vieną būdą. Kiekvienam asmeniui sudarysime po aibę, kur surašysime visus jo tiesioginius draugus. Tuomet sujungsime visas aibes, turinčias bendrų elementų. Jei pradžioje aibės a ir b bendrų elementų neturėjo, tačiau aibę a sujungus su kuria nors kita aibe, jų atsirado, tai aibes a ir b teks sujungti – jei asmuo C draugauja ir su A, ir su B, tai A paslaptys pasieks B, nors A ir B tarpusavyje nedraugauja.



const max = 100;   { maksimalus asmenų skaičius }

type  paslaib = array [1..max] of set of 1..max; 

                  { su kuo kiekvienas asmuo dalijasi paslaptimis }



function paslaptys (pasl: paslaib): integer;

                     { į aibę surašyti kiekvieno asmens draugai – }

                     { t.y. tie su kuriais jis paslaptimis dalijasi tiesiogiai }

  var paslsk, i, j: integer;

      turi: boolean; { ar aibės turi bendrų elementų }

begin

  { pirmiausia sujungiamos aibės, turinčios bendrų elementų }

  for i := 1 to max do

    begin

      turi := false;{ nerasta aibių, turinčių bendrų elementų su aibe i }

      for j := i + 1 to max do

        if pasl[i] * pasl[j] <> []

           then begin

                  pasl[j] := pasl[j] + pasl[i];

                  turi := true

                end;

      if turi then pasl[i] := []

    end;

  { liko aibės, neturinčios bendrų elementų }

  { paslapčių tiek, kiek liko netuščių aibių }

  paslsk := 0;

  for i := 1 to max do

    if pasl[i] <> []

       then paslsk := paslsk + 1;

  paslaptys := paslsk

end; { paslaptys }

103. Pašto ženklai. Sprendimą aiškinsime nagrinėdami sąlygoje pateiktą pavyzdį. Duotos vertės 1, 4, 12, 21, galima naudoti ne daugiau kaip 5 ženklus. Sudarome lentelę, kurioje tilptų visos galimos vertės (nuo 1 iki maxint). Joje žymėsimės kiek reikia panaudoti pašto ženklų, norint sudaryti atitinkamą vertę. Kol nesudarėme nei vieno rinkinio, lentelę užpildykime kokiu nors dideliu skaičiumi, pvz. 31. (tai yra didžiausias galimas pašto ženklų skaičius rinkinyje plius vienetas). 

0�1�2�3�4�5�6�7�8�9�10�11�12�13�...��0�31�31�31�31�31�31�31�31�31�31�31�31�31�...��

Į lentelę įterpėme papildomą vertę, lygią 0. Jai sudaryti reikia 0 pašto ženklų. Nuo tos vertės ir pradėsime. Prie nagrinėjamos vertės eilės tvarka pridėkime visas turimas vertes ir pažymėkime lentelėje, kad joms gauti prireikė vieno pašto ženklo. Gausime:

0�1�2�3�4�5�6�7�8�9�10�11�12�13�...��0�1�
31
�31�1�
31
�31�31�31�31�31�31�1�
31
�...��

Imame kitą vertę (vieną) ir vėl pridedame visas galimas vertes. Prie naujai gautų verčių pasižymime, kad joms gauti reikia 2 ženklų. 

Paėmę tolesnę vertę (du) ir pridėję turimų ženklų vertes, gausime tokią lentelę:

0�1�2�3�4�5�6�7�8�9�10�11�12�13�...��0�1�2�3�1�2�3�31�31�31�31�31�1�2�...��

Taip pildome lentelę, kol nagrinėjamai vertei gauti prireikia daugiau nei m ženklų. Įrašant ženklų skaičių į lentelę, būtina tikrinti, ar anksčiau gautas ženklų kiekis nebuvo mažesnis už dabartinį. Pavyzdžiui, pildydami paskutiniąją lentelę toliau, ir prie 3 pridėję 1, gautume, kad ketvertui sudaryti reikia jau 4 ženklų, nes trejetui gauti reikia 3 ženklų. Tačiau ketvertą jau buvome sudarę su mažesniu ženklų kiekiu – vienu ženklu. Todėl iš dviejų rinkinių renkamės mažesnįjį: prie ketverto paliksime vienetuką.



const MAX_N = 30; { maksimalus skirtingų pašto ženklų verčių skaičius }

      MAX_M = 30; { maksimalus ženklų kiekis rinkinyje }

type vertės = array [1..MAX_N] of integer;

     ženklai = array [0..maxint] of byte;

                        { jei pasirinktume integer tipą, pritrūktų atminties }



function didžiausia (n, m: integer; vertė: vertės)

                                               : integer;

  var ž: ženklai;

      i, j: integer;

begin

  for i := 0 to maxint do

    ž[i] := MAX_M;   { dar nesudarėme nei vieno ženklų rinkinio }

  i := 0;

  ž[0] := 0;

  while ž[i] <= m do { jei dar neviršijome ženklų limito rinkinyje }

    begin

      for j := 1 to n do

        if i + vertė[j] <= maxint

           then if ž[i + vertė[j]] > ž[i] + 1

                   then ž[i + vertė[j]] := ž[i] + 1;

      i := i + 1;

    end;

  didžiausia := i - 1;

  end; { didžiausia }

1 2 7 14 19 24 43 44 51 900�2684��5�25 26

1 5 6 9 17 30 38 52 54 57 73 94 112 127 158 166 168 204 214 263 300 364 427 490 521�12966��6�19 20

1 4 12 13 21 22 31 45 75 213 224 356 373 400 465 752 800 845 1000�18002��

Sunkiausia šiame uždavinyje – sugalvoti greitą algoritmą. Testai parinkti atsakymų didėjimo tvarka: jie tikrina sprendimų efektyvumą. 5 ir 6 testuose pašto ženklų vertės surašytos į dvi eilutes.



104. Lošimas degtukais. Žaidimo strategija paprasta – reikia stengtis imti tiek degtukų, kad priešininkas su likusiu degtukų skaičiumi pralaimėtų.

Jei likęs tik vienas degtukas, mes pralaimime (nes teks paimti paskutinį degtuką). Jei yra du degtukai, tuomet galime paimti vieną ir priešininkui palikti vieną. Jei turime tris degtukus, laimėti vėl nebegalime, nes leidžiama paimti tik vieną degtuką, o turėdamas du degtukus, priešininkas laimės. Turėdami keturis, penkis ar šešis degtukus, galime paimti tiek, kad priešininkas patektų į pralaimimą situaciją – jam liktų trys degtukai. 

Matome, kad jei duota 21 - 1, arba 22 - 1 ar 23 - 1 degtukų, situacija yra pralaimima. Žiūrime toliau. Antrajam žaidėjui teks pralaimima situacija, jei jam paliksime 23 - 1 degtukų. Tai galime padaryti, jei duotas degtukų skaičius yra iš intervalo [23, 24 - 2]. Visais šiais atvejais imdami ne daugiau pusės degtukų, galime priešininkui sudaryti pralaimimą situaciją. Jei duotas 24 - 1 degtukas, paėmus didžiausią galimą degtukų skaičių (24 - 1) div 2 ( 23, priešininkas paklius į laiminčią situaciją.

Dabar jau galime sudaryti žaidimo strategiją. Jei duotą degtukų skaičių galima užrašyti pavidalu n ( 2k - 1, tai situacija pralaimima, kitais atvejais situacija yra laimima ir reikia imti (n + 1) mod 2k, kur 2k yra didžiausias dvejeto laipsnis, neviršijantis n.



function degtukai (n: integer): integer;

  var dvej: integer;

begin

  dvej :=  1; { ieškome didžiausio dvejeto laipsnio, neviršijančio n }

  while (dvej <= n) do

    dvej := dvej * 2;

  dvej := dvej div 2;

  if (n + 1) mod dvej = 0

     then degtukai := 0

     else degtukai := (n + 1) mod dvej

end; { degtukai }

105. Traukinys. Tarkime, kad mokame suskaičiuoti keliais būdais galima suformuoti sąstatą, jei traukinį sudaro tik keleiviniai vagonai ir vagonai-restoranai. Jei traukinį sudaro n vagonų, tą skaičių pažymėkime f(n).

Dabar prisiminkime, kad yra dar dviejų rūšių vagonai. Pašto vagonų yra ne daugiau dviejų, o prekinių – ne daugiau trijų. 

Kai turime vieną pašto arba prekinį vagoną, tai yra du variantai – arba pašto vagonas priekyje arba prekinis gale (105.1.a-b pav.). 

105.1.a-b pav.

Kai yra du tokie vagonai, turime tris variantus – du pašto vagonai priekyje, vienas pašto priekyje ir vienas prekinis gale arba du prekiniai vagonai gale (105.2.a-c pav.). 

105.2.a-c pav.

Kai turime tris pašto ar prekinius vagonus yra trys variantai – du pašto ir prekinis, pašto ir du prekiniai arba trys prekiniai. 

Kai yra keturi tokie vagonai, galimi du variantai – du pašto ir du prekiniai arba vienas pašto ir trys prekiniai vagonai (105.3.a-b pav.). 

105.3.a-b pav.

Jei turime penkis pašto ir prekinius vagonus, yra vienas atvejis – du pašto ir trys prekiniai vagonai. 

Sąstate yra n vagonų ir jame gali būti vienas, du, trys, keturi, penki ar nei vieno prekinio ir pašto vagono. 

Jai sąstate yra vienas toks vagonas, tai jame yra n-1 prekinis ar keleivis vagonas. Skirtingų šių dviejų rūšių vagonų kombinacijų gali būti f(n-1). Kiekvieno tokio sąstato priekyje galima pridėti pašto vagoną arba gale prijungti prekinį vagoną. Taigi, variantų skaičius lygus 2 ( f(n-1).

Analogiškai samprotaudami gauname, kad kai n ( 6 iš viso yra 

f(n) + 2(f(n-1) + 3(f(n-2) + 3(f(n-3) + 2((n-4) + f(n-5) būdų sudaryti traukinį, čia f(n) rodo keliais variantais galima suformuoti traukinį, jei sąstatą sudaro n vagonų, o šie yra arba keleiviniai, arba restoranai.

Kai n mažesnis už 6, atmetami negalimi atvejai.



Belieka išsiaiškinti, kaip suskaičiuoti f(n). Vienas paprasčiausių būdų: traukinį formuoti rekursiškai. Tačiau toks algoritmas viršys laiko limitus. Galima pasinaudoti Dinaminio programavimo metodu. Susidarysime lentelę iš n elementų, ją pavadinkime v. Kiekvienas jos elementas sudarytas iš dviejų laukų. v[i].kele rodo keliais būdais galima suformuoti sąstatą iš dviejų rūšių vagonų, jei pirmasis vagonas yra keleivinis, v[i].rest – jei pirmasis vagonas yra restoranas. 

Mūsų ieškoma reikšmė yra v[n].kele, mat sąstatas negali prasidėti restoranu. Šios reikšmės ieškosime rekursiškai, tačiau kartu pildysime lentelę ir jei tos pačios reikšmės v[i] prireiks antrą ar trečią kartą, jos nebeieškosime rekursiškai, o pasiimsime iš lentelės.

Jei n ( 4, lentelė atrodytų taip:

�lauko kele reikšmė�Paaiškinimas�lauko rest reikšmė�Paaiškinimas��v[1]�1�K�0�Sąstatas negali baigtis restoranu��v[2]�1�KK�1�RK��v[3]�2�KKK

KRK�1�RKK��v[4]�3�KKKK

KRKK

KKRK�2�RKKK

RKRK��

const MAX_N = 43;

type vagonas = (kel, res);

     lentelė = array [1..MAX_N] of record

                                     kele, rest: longint;

                                   end;



function vagonai (n: integer): longint;

  var viso, i: longint;

      v: lentelė;

      { v[i].kele – keliais būdais galima suformuoti sąstatą iš i vagonų, }

      { jei sąstatas prasideda keleiviniu vagonu }

      { ir formuojamas tik iš restoranų ir keleivinių vagonų }



  function form (m: longint; { formuojamo sąstato ilgis }

                 pirm: vagonas { pirmojo vagono tipas }):

                                                 longint;

  { funkcijos rezultatas – v[m].kele reikšmė, jei pirm = kel ,

  { arba v[i].rest priešingu atveju }

  { šia reikšme funkcija papildo lentelę v }

    var kiek: longint;

  begin

    { gal būt reikiamos reikšmės jau apskaičiuotos }

    if (pirm = kel) and (v[m].kele <> -1)

    then form := v[m].kele

    else if (pirm = res) and (v[m].rest <> -1)

         then form := v[m].kele

         else begin

                { po keleivinio vagono gali eiti arba keleivinis }

                if v[m-1].kele = -1

                   then kiek := form (m-1, kel)

                   else kiek := v[m-1].kele;

                { arba vagonas-restoranas }

                if (m >= 3) and (pirm = kel)

                then if v[m-1].rest = -1

                     then kiek := kiek + form (m-1, res)

                     else kiek := kiek + v[m-1].rest;

                { papildome lentelę v }

                if pirm = kel

                   then v[m].kele := kiek

                   else v[m].rest := kiek;

                form := kiek;

              end;

  end; { form }



begin

  for i := 1 to n do

    begin

      v[i].kele := -1;  { sąstatai dar nesuformuoti }

      v[i].rest := -1;

    end;

  v[1].kele := 1;

  v[1].rest := 0; { sąstatas negali prasidėti restoranu }

  if n > 1

     then viso := form (n, kel)

     else viso := 1;

  if n > 1 then viso := viso + 2 * v[n-1].kele;

  if n > 2 then viso := viso + 3 * v[n-2].kele;

  if n > 3 then viso := viso + 3 * v[n-3].kele;

  if n > 4 then viso := viso + 2 * v[n-4].kele;

  if n > 5 then viso := viso + v[n-5].kele;

 vagonai := viso

end; { vagonai }



Yra ir kitas, nors matematiškas, tačiau labai įdomus būdas suskaičiuoti f(n). Išvesime rekurentinę formulę kaip skaičiuoti f(n), žinant ankstesnes reikšmes f(1), f(2), ...f(n-1). Naujus vagonus bandysime prijungti iš priekio. Pirmasis naujo sąstato vagonas yra keleivinis. Antrasis gali būti arba keleivinis arba restoranas. 

Tarkime, kad n ( 5.

Imame visas galimas kombinacija iš keturių vagonų (105.4.a-c pav.):

105.4.a-c pav.

Prie kiekvieno tokio sąstato priekyje galime prijungti keleivinį vagoną ir gausime naują sąstatą iš n vagonų (105.5.a-c pav.).

105.5.a-c pav.



Taip gavome visas įmanomas kombinacijas, kurių antrasis vagonas yra keleivinis.

Taigi, f(n) ( f(n-1) + ?;

Norint gauti visas kombinacijas, kur sąstato antrasis vagonas yra restoranas, reikia imti visus įmanomus sąstatus, sudarytus iš n-2 vagonų, prasidedančius keleiviniu vagonu. Tokių sąstatų yra f(n-2). Imkime mūsų pavyzdį (105.6.a-b pav.):

105.6.a-b pav.



Prie jų prijungiame po du vagonus (105.7.a-b pav.):

105.7.a-b pav.



Taigi, f(n) ( f(n-1) + f(n-2). Kadangi f(1) ( f(2) ( 1, gauname, kad f(n) lygus �n-ąjąm Fibonačio skaičiui.



const MAX_N = 43; { maksimalus vagonų skaičius }

type lentelė = array [1..MAX_N] of longint;



function vagonai (n: longint): longint;



  procedure fibonači (n: longint; var fibo: lentelė);

  { sudaro fibonačio skaičių nuo 1 iki n lentelę }

    var i: longint;

  begin

    fibo[1] := 1; fibo[2] := 1;

    for i := 3 to n do

      fibo[i] := fibo[i-1] + fibo[i-2];

  end; { fibonači }



  var kiek: longint;

      fibo: lentelė;

  begin

  fibonači (n, fibo);

  kiek := 0;

  if n > 5 then kiek := kiek + fibo[n-5];

  if n > 4 then kiek := kiek + 2 * fibo[n-4];

  if n > 3 then kiek := kiek + 3 * fibo[n-3];

  if n > 2 then kiek := kiek + 3 * fibo[n-2];

  if n > 1 then kiek := kiek + 2 * fibo[n-1];

  kiek := kiek + fibo[n];

  vagonai := kiek

end; { vagonai }

106. Loginis reiškinys. Šis uždavinys labai primena 98 uždavinį „Reiškinys“. Sprendimas būtų labai panašus. 

Reiškinį nagrinėjame nuo pradžios po vieną simbolį. Atskiriame skaičius, kintamuosius, logines bei lyginimo operacijas. Visus skaičius, kintamuosius bei operacijas įsimename, o pabaigę nagrinėti reiškinį, suskaičiuojame kiek buvo skirtingų skaičių, kintamųjų bei operacijų.

Sprendžiant šį uždavinį reikia atkreipti dėmesį į du dalykus:

1)  kintamųjų vardai gali būti užrašyti ir mažosiomis ir didžiosiomis raidėmis:  sk1 or SK1, todėl verta visas raides reiškinyje pakeisti didžiosiomis;

2)  yra vienintelis atvejis, kai dvi operacijos gali būti viena šalia kitos: tai operacija and arba or ir operacija not; operacija not gali prasidėti ir pats reiškinys:

not a or b

a or not b

Beje, reiškinyje gali būti skliaustų, kurie iš esmės nekeičia uždavinio sprendimo. Juos paprasčiausiai galima pakeisti tarpais, pavyzdžiui, jei turime reiškinį:

(a <( b) or (cd <> 6) and ((-14) < 13))?

tai pašalinę skliaustus, gauname:

a <( b or cd <> 6 and -14 < 13?

Kiek kebliau yra su ženklais ‘+’ ir ‘-‘, kurie gali eiti tiek prieš neigiamus skaičius, tiek prieš kintamuosius:

(-5 < 6) or (-a <( b)

Į šiuos ženklus galima iš viso nekreipti dėmesio, išskyrus vienintelį atvejį – kai prieš skaičių parašytas minusas. Tuomet jį reikia įsidėmėti kartu su skaičiumi.



const max = 200;

      max_ilg = 20; { maksimalus kintamojo vardo ar skaičiaus ilgis }

type lentelė = array [1..max] of string[max_ilg];

     loginės = (ne, ir, arba);

     lyginimo = (lygu, nelygu, daugiau, mažiau,

                 dauglygu, mažlygu);

     log_aibė = set of loginės; { loginių operacijų aibė }

     lyg_aibė = set of lyginimo; { lyginimo operacijų aibė }



procedure tarp_plius (var s: string);

{ pašalina tarpus ir ženklą '+' iš reiškinio s pradžios }

begin

  while (s[1] = ' ') or (s[1] = '+')do

    delete (s, 1, 1)

end; { tarp_plius }



procedure skaičius (s: string; var ilgis: integer);

{ s – reiškinys, prasidedantis skaičiumi, procedūra randa jo ilgį }

begin

  { ši procedūra sutampa su analogiška 98 užd. procedūra }

end; { skaičius }



procedure kintamasis (s: string; var ilgis: integer);

{ s – reiškinys, prasidedantis kintamuoju, procedūra randa jo ilgį }

{ ši procedūra sutampa su analogiška 98 užd. procedūra }

begin

  ...

end; { kintamasis }



procedure operacija(s: string;  var ilgis: integer;

                     var login: log_aibe; var lygin: lyg_aibė);

{ s – reiškinys, prasidedantis operacija }

{ procedūra randa pačią operaciją ir jos ilgį }

  var op: lyginimo;

begin

  case s[1] of

    'N': begin              { not }

            ilgis := 3;

            login := login + [ne];

          end;

    'A': begin               { and }

            ilgis := 3;

            login := login + [ir];

          end;

    'O': begin              { or }

            ilgis := 2;

            login := login + [arba]

          end;

     else if s[2] in ['>', '=']   { <>, <=, >= }

          then begin

                 ilgis := 2;

                 if s[1] = '>'

                 then op := dauglygu

                 else if (s[1] = '<') and (s[2] = '>')

                         then op := nelygu

                         else op := mažlygu;

                 lygin := lygin + [op];

               end

          else begin          { +, <, > }

                 ilgis := 1;

                 if s[1] = '='

                 then op := lygu

                 else if s[1] = '<'

                         then op := mažiau

                         else op := daugiau;

                 lygin := lygin + [op];

               end

  end

end; { operacija }



procedure kiek_skirt (skaič, kintam: lentelė;

                      login: log_aibė;

                      lygin: lyg_aibė;

                      var sk, kint, lyg, log: integer);

{ randa skirtingų skaičių, kintamųjų bei operacijų kiekį }

  const epsilon = 0.1e-10; { realieji skaičiai lyginami tokiu tikslumu }

  var ly: lyginimo;

      lo: loginės;

      a, b, kodas, kiek, i, j: integer;

      lygus, real_a, real_b: boolean;

      areal, breal: real;

begin

  { rasime kiek yra skirtingų skaičių }

  kiek := 0;

  i := 1;

  while i <= sk do

    begin

      { įtraukiame i-ąjį skaičių }

      kiek := kiek + 1;

      j := i + 1;

      while j <= sk do { randame skaičius, sutampančius su i-tuoju }

        begin

          { nustatysime, ar skaičiai yra vienodų tipų }

          real_a := (pos('.', skaič[i]) > 0) or 

                    (pos('E', skaič[i]) > 0);

          real_b := (pos('.', skaič[j]) > 0) or

                    (pos('E', skaič[j]) > 0);

          lygus := false;

          if real_a and real_b

          then begin

                 val(skaič[i], areal, kodas);

                 val(skaič[j], breal, kodas);

                 lygus := abs(areal-breal) < epsilon;

               end

          else if not real_a and not real_b

                  then begin

                         val(skaič[i], a, kodas);

                         val(skaič[j], b, kodas);

                         lygus := a = b

                       end;

            if lygus { panaikiname j-ąjį skaičių }

            then begin

                   skaič[j] := skaič[sk];

                   sk := sk - 1;

                 end

            else j := j + 1

        end;

      i := i + 1

    end;

  { rasime skirtingų kintamųjų skaičių }

  kiek := 0;

   for i := 1 to kint do

     if kintam[i] <> ''  { jei šis kintamasis dar nepriskaičiuotas }

        then begin

               kiek := kiek + 1;

               for j := i + 1 to kint do

                 if kintam[i] = kintam[j]

                    then kintam[j] := ''

             end;

  kint := kiek;

  { rasime skirtingų lyginimo operacijų skaičių }

  lyg := 0;

  for ly := lygu to mažlygu do

    if ly in lygin

       then lyg := lyg + 1;

  { rasime skirtingų loginių operacijų skaičių }

  log := 0;

  for lo := ne to arba do

    if lo in login

       then log := log + 1

end; { kiek_skirt }



procedure skaičiuoti (s: string; { pradinis reiškinys }

                      var sk, kint, lyg, log: integer);

{ suskaičiuoja kiek reiškinyje yra skaičių, kintamųjų ir operacijų }

  var i, ilgis: integer;

      skaič, kintam: lentelė;

      login: log_aibė;

      lygin: lyg_aibė;

begin

  { visos mažosios raidės reiškinyje keičiamos didžiosiomis }

  for i := 1 to length(s) do

    s[i] := upcase (s[i]);

  { jei tarp ženklo '-' ir skaičiaus yra tarpas (ai), tie tarpai panaikinami, jei po '-' }

  { eina kintamasis, tuomet panaikinamas pats ženklas }

  i := 1;

  while i <= length(s) do

    if s[i] = '-'

       then begin

              while s[i+1] = ' ' do

                delete (s, i+1, 1);

              if not (s[i + 1] in ['0'..'9'])

                 then delete (s, i, 1)

                 else i := i + 1

            end

       else i := i + 1;

    end;

  { skliaustai pakeičiami tarpais }

  for i := 1 to length(s) do

    if s[i] in ['(', ')']

       then s[i] := ' ';

  tarp_plius (s); { pašalinami tarpai ir ženklas '+' reiškinio priekyje }

  sk := 0;

  kint := 0;

  lygin := [];

  login := [];

  while s <> '?' do  { kol neišnagrinėjome viso reiškinio }

    begin

      { reiškinys prasidės arba skaičiumi, arba kintamojo vardu }

      if s[1] in ['0'..'9', '-']

         then begin

                skaičius (s, ilgis);

                sk := sk + 1;

                skaič[sk] := copy (s, 1, ilgis);

              end

         else begin

                kintamasis (s, ilgis);

                { reikia patikrinti, ar tai tikrai ne operacija not }

                if copy (s, 1, ilgis) <> 'NOT'

                then begin

                       kint := kint + 1;

                       kintam[kint] := copy (s, 1, ilgis)

                     end

                else ilgis := 0

              end;

      { pašaliname tą skaičių ar kintamąjį }

      delete (s, 1, ilgis);

      tarp_plius (s);

      { jei tai ne reiškinio pabaiga, randame operaciją }

      if s <> '?'

         then begin

                operacija (s, ilgis, login, lygin);

                delete (s, 1, ilgis)

              end;

      tarp_plius (s)

    end;

  kiek_skirt(skaič, kintam, login, lygin, sk, kint, lyg, log)

end; { skaičiuoti }

Aukščiau pateiktas algoritmas yra gan ilgas. Jį galima gerokai sutrumpinti, bet algoritmas taps sunkiau suprantamas. Todėl palikome ilgesnį sprendimą.



Testo nr.�Pradiniai duomenys�Rezultatai��1�(5 < 3) and (6 < 4) or not (1 < 2) and (3 > 2)?�6 0 2 3��2�(5<3)or(6>4)or(7=5)or(8<=10)or(9>=11)or(9<>11) or(10=8)?�9 0 6 1��3�ora or nota or anot?�0 3 0 1��4�sk1 or SK1 or Sk1 or sK1?�0 1 0 1��5�a and b or c and ab or cb or ab or ba or abc or cba or aba?�0 9 0 2��6�< 5) or (0.5E01 < 6) or (0.5e01 < 7) or 

(0.0005E+4 < 5) or (50e-1 < 5) or (0.5 < 50)?�6 0 1 1 ��7�(-2 < 2) or (-a <> b) or (7 > -a4)?�3 3 3 1��

Vyresniems (X – XII kl.)

107. Žaidimas. Tai yra klasikinis perrinkimo (grįžimo metodo) uždavinys. Jo sprendimas gerai išnagrinėtas [3]. 

Tarkime, kad turime kokią nors situaciją žaidimo lentoje. Pasirenkame bet kurį ėjimą, jį atliekame, įsimename ir taip žaidžiame, kol susidaro situacija lentoje, kai nebeįmanoma nieko nukirsti. Jei lentoje liko tik viena šaškė, sprendinys rastas, jei ne – atstatome paskutinį įsimintą ėjimą ir bandome eiti kitaip. Jei išbandome visus ėjimus ir nei vienas netinka, grįžtame dar per žingsnį atgal. 

Darbą baigiame, kai randame sprendinį arba išbandome visus galimus variantus.



Algoritmo sudarymui paimsime 
d
vimat
į
 
kvadratinį 
masyvą
, kuriame tilptų lenta o kraštiniai masyvo elementai liktų už lentos ribų. 
Lentą uždėsime ant kvadrato taip, kad jų centriniai langeliai sutaptų. 
Masyvo elementus, atitinkančius 
lentos 
langelius su šaškėmis užpildysime 
vienetais (
1
)
, o atitinkančius 
tuščius langelius 
– nuliais (0). Masyvo langelius, neįeinančius į lentą taip pat užpildysime 
minus vieneta
i
s
 (
-1
)
.


const MAX_N = 5;

  MAX_M = 5;

  MAX = MAX_N div 2 + MAX_M+2; { maksimalaus kvadrato, kuriame telpa }

            
          
{ lenta, kraštinė
; lenta apgaubiama tuščiais langeliais}

  MAX_K = 150; { maksimalus kirtimų skaičius }

type t_lenta = array [-MAX..MAX, -MAX..MAX] of integer;

                             { 1 – šaškė, 0 – tuščia -1 – už lentos ribų }

     kirtimai = array [0..MAX_K] of 

                        record

                          x1, y1, x2, y2: integer

                        end;



procedure šaškės (dydis, { 
n
 
div
 2 + 
m
 
+
 
2
}

                  ša: integer; { šaškių skaičius lentoje }

                  lenta: t_lenta;

                  var nuk: kirtimai; { kirtimų seka }

                  var rasta: boolean { ar rastas sprendinys });

  var k: integer;     { nukirstų šaškių skaičius }



  procedure kirsti (x1, y1, x2, y2: integer);

  { rekursinė procedūra, atliekanti kirtimus }

    var i, j: integer;

  begin

    k := k + 1;  { įsidėmime kirtimą }

    nuk[k].x1 := x1;     nuk[k].x2 := x2 + (x2 - x1); 

    nuk[k].y1 := y1;     nuk[k].y2 := y2 + (y2 - y1);

    lenta[x1, y1] := 0; 

    lenta[x2, y2] := 0;

    lenta[x2 + (x2 - x1), y2 + (y2 - y1)] := 1;

    if k = ša - 1

       then rasta := true



    
else
 



 
    for i := -dydis to dydis do

 
 
     for j := -dydis to dydis do

  
 
      if (lenta[i, j] = 1) and not rasta

   
 
        then if (lenta[i,j+1]=1) and (lenta[i,j+2]=0)

    
 
                then kirsti (i, j, i, j+1)

     
 
      else if (lenta[i,j-1]=1) and (lenta[i,j-2]=0)

      
 
              then kirsti (i, j, i, j-1)

       
 
    else if (lenta[i-1,j]=1) and (lenta[i-2,j]=0)

        
 
            then kirsti (i, j, i-1, j)

         
 
  else if (lenta[i+1,j]=1) and (lenta[i+2,j]=0)

          
 
         then kirsti (i, j, i+1, j);

    { grąžiname nukirstą šaškę į vietą }

    if not rasta

       then begin

              k := k - 1;

              lenta[x1, y1] := 1;

              lenta[x2, y2] := 1;

              lenta[x2 + (x2 - x1), y2 + (y2 - y1)] := 0;

            end;

  end; { kirsti }



begin  { šaškės }

  rasta := false;

  k := -1;

  { pirmasis kirtimas bus fiktyvus }

  kirsti (dydis, dydis, dydis, dydis)

end; { šaškės }



Šiam uždaviniui pateikti 6 testai. Jie visi pavaizduoti 107.2.1-6 paveikslėliuose. Atsakymų pavyzdžiai pateikti kartu su pradinių duomenų bylomis, tačiau dauguma testų turi ne vieną sprendinį. Antro ir ketvirto testų atvejais atsakymas yra NEGALIMA.

107.2.1-6 pav.



108. Nesėkmingas skrydis oro balionu. Norint išspręsti šį uždavinį, reikia žinoti šiek tiek geometrijos. Idėja labai paprasta: iš taško, kuriame nusileido daugiakampis brėžiame spindulį bet kuria kryptimi. Suskaičiuojame kiek kartų spindulys kerta daugiakampį. Jei lyginį kartų skaičių – balionas nusileido valstybės teritorijoje, jei nelyginį – už jos ribų (108.1 pav.). 

108.1 pav.



Taigi, imame visas daugiakampio briaunas ir skaičiuojame kelios iš jų kerta spindulį.

Reikia nepamiršti keleto atskirų atveju:

a) spindulys yra lygiagretus daugiakampio kraštinei ir ją liečia (108.2 pav.):

108.2 pav.

tada situacijoje spindulys daugiakampio nekerta.

b) spindulys kerta daugiakampio viršūnę (108.3.a-b pav.):

108.3.a-b pav.

Pirmuoju atveju kirsdamas viršūnę, spindulys tik liečia daugiakampį, antruoju – kerta. Tarkime, kad kertamos viršūnės numeris yra i. Tuomet reikia paimti i-1-ąją ir i+1-mąją viršūnes ir patikrinti ar jos yra toje pačioje tiesės (gautos pratęsus spindulį) pusėje.

Remdamiesi šia idėja parašome algoritmą.

const MAX_N = 1000; { maksimalus daugiakampio viršūnių skaičius }

      type taškas =  record

                       x, y: -1..1000;

                     end;

           daugiak = array [1..MAX_N] of taškas;



function succ_v (i, n: integer): integer;

begin

  if i = n

     then succ_v := 1

     else succ_v := i + 1

end; { succ_v }



function pred_v (i, n: integer): integer;

begin

  if i = 1

     then pred_v := n

     else pred_v := i - 1

end; { succ_v }



function viduje (v: daugiak; n: integer; { daugiakampis }

                 t: taškas { keliautojų koordinatės }): boolean;

  var i,

      kiek: integer;  { kiek briaunų kerta spindulį }

      br1, br2, s1, s2: taškas;

      prikl: boolean; { ar taškas priklauso daugiakampio briaunai }

begin

  { patikrinsime, ar taškas priklauso daugiakampio briaunai }

  prikl := false;

  for i :=  1 to n do

    prikl := prikl or priklauso(v[i], v[succ_v(i,n)], t);

  if prikl   { jei keliautojai nusileido ant sienos }

  then viduje := true

  else       { priešingu atveju }

    begin

      s1.x := -1; s1.y := -1; { spindulys tikrai išeis už } 

      s2 := t;                { daugiakampio ribų }

      kiek := 0;

      for i := 1 to n do { skaičiuosime kiek kraštinių kerta spindulį }

        begin

          br1 := v[i];       { imame kitą briauną }

          br2 := v[succ_v (i, n)];

          if not lygiagr (s1, s2, br1, br2)

             { jei  briauna nelygiagreti spinduliui }

             then if virš (s1, s2, i, n, v)

             { spindulys kerta (o ne liečia) daugiakampį  viršūnėje}

             or kerta (s1, s2, br1, br2) { arba kerta briauną }

             then kiek := kiek + 1;

        end;

      viduje := kiek mod 2 = 1

    end

end; { viduje }



Dabar galime detalizuoti procedūras. Tarkime, kad tiesė nusakoma dviem taškas t1(x1, y1) ir t2 (x2, y2) ir turime trečią tašką t (x, y). Iš geometrijos žinome kad, jei :



                                      < 0,     taškas t yra kairėje tiesės pusėje;

(x - x1)     (    y - y1         ( 0       taškas t priklauso tiesei;                            (1)

(x2 - x1)         y2 - y1         > 0       taškas t yra dešinėje tiesės pusėje;

jei tiesės kryptis yra iš taško t1 į t2.

Tikrinant, ar keliautojai nusileido ant sienos reikia patikrinti, ar taškas priklauso atkarpai. Taip bus jei taškas priklausys tiesei, gautai pratęsus atkarpą ir jei taško koordinatės bus apribotos atkarpos galų koordinatėmis:

function priklauso (t1, t2, t: taškas): boolean;

{ ar taškas t priklauso atkarpai, išvestai per taškus t1 ir t2 }

begin

  priklauso := ((t.x-t1.x) * (t2.y-t1.y) - { priklauso tiesei }

                (t.y-t1.y) * (t2.x-t1.x) = 0) and

            ((t1.x <= t.x) and (t.x <= t2.x) or { ir atkarpai }

             (t2.x <= t.x) and (t.x <= t2.x)) and

            ((t1.y <= t.y) and (t.y <= t2.y) or

             (t2.y <= t.y) and (t.y <= t2.y))

end; { priklauso }



Dvi tiesės, išvestos per taškus (t1, t2) ir (t3, t4) yra lygiagrečios, jei:

(x2 - x1)     (    x4 - x3 

(y2 - y1)           y4 - y3                                                                          (2)



function lygiagr (t1, t2, t3, t4: taškas): boolean;

{ ar tiesė, išvesta per taškus t1 ir t2 lygiagreti tiesei }

{ išvestai per kitus du taškus }

begin

  lygiagr := (t2.x - t1.x) * (t4.y - t3.y) =

             (t2.y - t1.y) * (t4.x - t3.x)

end; { lygiagr }



Pereiname prie funkcijos virš. Ar i-toji viršūnė priklauso spinduliui, galima nustatyti pasinaudojant funkcija priklauso. Jei taip, imame dvi gretimas viršūnes (i-1-mą ir iž1-mą) ir patikriname ar jos abi yra priešingose tiesės (gautos pratęsus spindulį) pusėse. Jei taip – daugiakampis kertamas, jei ne – liečiamas. Patikrinti galima pagal (1) formulę:

function dešinė (t1, t2, t: taškas): boolean;

{ ar taškas t yra dešinėje tiesės išvestos per taškus }

{ t1 ir t2 pusėje. Tiesės kryptis iš t1 į t2 }

begin

  dešinė := (t.x - t1.x) * (t2.y - t1.y) -

            (t.y - t1.y) * (t2.x - t1.x) > 0

end; { dešinė }



function virš (s1, s2: taškas; i, n: integer; v: daugiak)

                                               : boolean;

{ tikrina, ar i-toji viršūnė priklauso spinduliui ir jei taip, }

{ ar spindulys šioje viršūnėje kerta (o ne liečia) daugiakampį }

begin

  if priklauso (s1, s2, v[i]) { jei taškas priklauso spinduliui }

     then virš := dešinė (s1, s2, v[succ_v(i, n)]) <>

                  dešinė (s1, s2, v[pred_v(i, n)])

     else virš := false

end; { virš }



Liko patikrinti, ar spindulys kerta (o ne liečia!) atkarpą. Atrodytų, kad tai gan paprasta: spindulį pratęskime iki tiesės ir patikrinkime, ar abu atkarpos galai yra priešingose tiesės pusėse. Tačiau jei tiesė kerta atkarpą, dar nebūtinai spindulys ją kirs (žr 69.3 pav.). Todėl reikia pratęsti atkarpą iki tiesės, parinkti tikrai daugiakampiui nepriklausantį spindulio tašką ir patikrinti ar spindulio pradžia ir pasirinktasis taškas tikrai yra priešingose tiesės pusėse. 

T. y. faktiškai paimamos dvi atkarpos ir tikrinama, ar jos nesikerta. 

function kerta (a1, a2, b1, b2: taškas): boolean;

{ ar atkarpa |a1, a2| kerta atkarpą |b1, b2| }

{ jei atkarpos nėra lygiagrečios }

begin

  kerta := (dešinė(a1, a2, b1) <> dešinė(a1, a2, b2)) and

           (dešinė(b1, b2, a1) <> dešinė(b1, b2, a2)) and

           not priklauso(a1, a2, b1) and 

           not priklauso(a1, a2, b2)

end; { kerta }





Šiam uždaviniui parengta 20 testų. Testai suskirstyti į penkias grupes po keturis testus. Kiekvienoje grupėje pateikiamas tas pats daugiakampis, tačiau keturi skirtingi nusileidimo taškai. Daugiakampiai turi pakankamai daug viršūnių, todėl daugiakampius nupiešėme, bet neįdėjome jų viršūnių sąrašų (108.4.1-4 pav.). 

108.4.1-4 pav.



Grupės nr. �Testo nr. grupėje�Nusileidimo

taško koordinatės�Rezultatas�Taškas yra��1�1�132 89�TAIP�ant horizontalios briaunos���2�140 84�TAIP�kampe tarp dviejų briaunų��Iškilas �3�129 83�TAIP�Viduje��daugiakampis�4�128 82�NE�Išorėje��2�1�157 103�TAIP�ant vertikalios briaunos��Neiškilas �2�160 99�TAIP�žvaigždės viduryje��žvaigždės formos�3�156 98�NE�išorėje��daugiakampis�4�165 98�TAIP�žvaigždės kampe��3�1�12 12�NE�tarp dviejų smaigalių��Dantytai kirpto �2�9 12�TAIP�kairėje dviejų smaigalių��lapo formos �3�9 11�NE�tarp dviejų dantų��daugiakampis�4�5 6�TAIP�parinktas atsitiktinai��4�1�6 6�TAIP�Jei leidžiamas vertikalus ar��Neiškilusis �2�8 6�TAIP�horizontalus spindulys, šis testas��daugiakampis�3�6 9�NE�tikrina atskirus atvejus (kai ���4�6 8�TAIP�spindulys liečia kampus ir pan.)��5�1�77 87�TAIP���Spiralės formos�2�78 86�NE���neiškilusis�3�81 81�TAIP���daugiakampis�4�77 81�NE���

110. Atspėk skaičių. Tai pirmasis dialoginis uždavinys, pateiktas Lietuvos Informatikos olimpiadose. Jo sprendimas nėra sudėtingas. Svarbu, kad programa tinkamai pasinaudotu pateiktu moduliu.

Sugalvotasis skaičius gali būti bet koks keturženklis skaičius iš intervalo [1000; 9999], kurio visi skaitmenys skirtingi. Taigi, susidarome visų keturženklių skaičių kurių skaitmenys skirtingi sąrašą. Sąraše saugosime galimus sprendinius.

Pirmąjį spėjamą skaičių parenkame atsitiktinai – bet kurį skaičių iš sąrašo. Tarkime, kad tai buvo 4521. Gaunamas atsakymas: atspėti du skaičiai, iš jų vienas yra savo vietoje. Pasinaudokime šia informacija. Sąraše palikime tik tuos skaičius, kurių lygiai du skaitmenys sutampa su spėtuoju skaičiumi ir iš dviejų tik vienas yra toje pačioj vietoje, kaip ir spėtajame skaičiuje. Pavyzdžiui, skaičiai 4302, 1589 bus palikti sąraše, o skaičiai 4891, 1284 pašalinti.

Toliau vėl spėjama atsitiktinai. Imamas bet kuris sąraše likęs skaičius, ir vėl pasinaudojama pateiktu atsakymu. Veiksmai kartojami, kol sąraše lieka tik vienas skaičius. 

Spėjant tokiu būdu bet kokį skaičių įmanoma atspėti per 7 ar mažiau kartų. Todėl laikoma, kad atsakymas gautas, jei atspėta per 7 ar mažiau kartų.

program atspėk_skaičių;

  uses žaidimas;

  type sąrašas = ^elem;

       elem = record

                ska: 1000..9999;

                tolesnis: sąrašas

              end;



  function skirtingi (ket: integer): boolean;

  { ar keturženklio skaičiaus ket skaitmenys yra skirtingi }

    var aibė: set of char;

        elem_sk: integer;

        c: char;

        eil: string;

  begin

    str (ket, eil);  { skaičius paverčiamas eilute }

    aibė := [eil[1]] + [eil[2]] + [eil[3]] + [eil[4]];

    elem_sk := 0;

    for c := '0' to '9' do

      if c in aibė

         then elem_sk := elem_sk + 1;

    skirtingi := elem_sk = 4;

  end; { skirtingi }



  procedure sąrašo_sudarymas (var s: sąrašas);

    var p, p1: sąrašas;

        ket: integer;

  begin

    new (p);

    p^.ska := 1023; { mažiausias keturženklis skaičius, kurio }

                    { visi skaitmenys skirtingi }

    p1 := p;

    for ket := 1024 to 9999 do

      if skirtingi (ket)

         then begin   { įtraukiame skaičių į sąrašą }

                new (p^.tolesnis);

                p := p^.tolesnis;

                p^.ska := ket

              end;

    p^.tolesnis := nil;

    s := p1

  end; { sąrašo_sudarymas }



  function vienodi (sk1, sk2: integer): integer;

  { suskaičiuoja, kiek skaičiai sk1 ir sk2 turi vienodų skaitmenų }

    var aibė: set of 0..9;

        kiek: integer;

  begin

    aibė := [];

    while sk1 <> 0 do { pirmojo skaičiaus skaitmenis įrašome }

      begin           { į aibę }

        aibė := aibė + [sk1 mod 10];

        sk1 := sk1 div 10;

      end;

    kiek := 0;

    while sk2 <> 0 do   { tikriname kiek antrojo skaičiaus skaitmenų }

      begin             { yra aibėje }

        if sk2 mod 10 in aibė

           then kiek := kiek + 1;

        sk2 := sk2 div 10

      end;

    vienodi := kiek

  end;  { vienodi }



  function vietoje (sk1, sk2: integer): integer;

  { suskaičiuoja, kiek skaičiai sk1 ir sk2 turi vienodų skaitmenų }

  { esančių tose pačiose pozicijose }

    var eil1, eil2: string[4];

        i, kiek: integer;

  begin

    str (sk1, eil1);

    str (sk2, eil2);

    kiek := 0;

    for i := 1 to 4 do

      if eil1[i] = eil2[i]

         then kiek := kiek + 1;

    vietoje := kiek

  end; { vietoje }



  procedure spėti (s: sąrašas);

    var sk, atsp, vietoj: integer;

        p: sąrašas;

  begin

    while s <> nil do

      begin { kol sąraše daugiau nei vienas elementas }

        sk := s^.ska;

        ėjimas (sk, atsp, vietoj);

        p := s;  { peržiūrėsime sąrašą }

        while p^.tolesnis <> nil do

          begin

            if (vienodi(sk, p^.tolesnis^.ska) <> atsp) or

               (vietoje(sk, p^.tolesnis^.ska) <> vietoj)

               then { pašalinamas elementas }

                    p^.tolesnis := p^.tolesnis^.tolesnis

               else p := p^.tolesnis

          end;

        s := s^.tolesnis; { pašalinamas pirmasis sąrašo elementas }

      end;                { nes jį jau spėjome }

    ėjimas (sk, atsp, vietoj);

  end; { spėti }



  var s: sąrašas;

begin

  sąrašo_sudarymas (s);

  spėti (s);

end.



Rodyklės tipas yra sudėtinga duomenų struktūra. Dėl to pailgėjo ir pati programa. Juk keturženklius skaičius buvo galima saugoti ir loginiame masyve. Rodyklės tipą pasirinkome nes šis tipas sudaro daug sunkumų sprendžiantiems: norėjome pateikti dar vieną algoritmo, kuriame vartojami šie duomenų tipai, pavyzdį. O šiame uždavinyje jį vartoti logiška: jei tik skaičius nėra tinkamas, jis pašalinamas iš sąrašo ir daugiau nenagrinėjamas. 

Pastaba. Sąlygoje pateikėme demonstracinį modulį žaidimas. Testavimui vartotas modulis iš esmės nesiskyrė nuo demonstracinio. Jame buvo įdėta papildoma apsauga nuo bandančių nesąžiningai spręsti uždavinį. 

Testo nr.�Pradinis duomuo�Aukščiau pateiktas algoritmas atspėjo per tiek kartų��1�7536�7��2�1598�5��3�4628�5��4�3210�4��5�6809�6��6�3589�4��7�8129�5��8�2831�7��9�4567�2��10�9182�7��

111. Tiesiami keliai. Toks uždavinys vadinamas Maksimalių srautų uždaviniu. 

Maksimalių srautų uždavinys yra vienas lengviausių iš Srautų tinklų uždavinių klasės, sprendžiamas Srautų leidimo žemyn (Preflow-push) metodu.

Maksimalių srautų uždavinys

Turime orientuotą grafą, kurį sudaro viršūnių ir briaunų aibė. Grafas yra svorinis, t. y. kiekviena briauna (u, v) turi neneigiamą svorį sv(u, v) > 0. Tą svorį vadinkime briaunos pralaidumu. Dvi grafo viršūnės išsiskiria iš kitų viršūnių: tai pradinė viršūnė p ir galinė viršūnė g. Susitarkime, kad grafas pasižymi tokia savybe: iš pradinės viršūnės p galima pasiekti bet kurią kitą viršūnę ir iš bet kurios viršūnės galime pasiekti galinę viršūnę g.

Toks grafas gali modeliuoti ne tik kelių žemėlapį. Tai gali būti ir elektros grandinė ar namo šildymo sistema. Pavyzdžiui, į namą ateina storas vamzdis su karštu vandeniu ir išsišakoja į storesnius ir plonesnius vamzdžius. Tie vamzdžiai išvedžiojami po namą, tai sujungiami, tai vėl atskiriami įvairiuose aukštuose, kol ataušęs vanduo surenkamas kitame vamzdyje. Žinomi visų vamzdžių pralaidumai. Koks didžiausias kiekis vandens gali ateiti į storąjį vamzdį. 

Srautai s yra apibrėžti visoms grafo briaunoms. Jie pasižymi šiomis savybėmis:

1) s(u, v) ( sv(u, v). Tai reiškia, kad briauna praleidžiami srautai negali viršyti briaunos pralaidumo. Pavyzdžiui, jei siauroje gatvelėje tiesiamas kelias ir jame tilps tik dvi juostos, negalima leisti mašinoms važiuoti trimis juostomis.

2) s(u, v) ( -s(v, u). Į viršūnę ateinantys srautai žymimi teigiamu skaičiumi, iš jos išeinantys – neigiamu. 

3) Kiekvienai viršūnei, išskyrus p ir g, galioja taisyklė: Į ją ateinančių srautų suma lygi iš jos išeinančių srautų sumai. Jei taisyklės nebūtų laikomasi galėtų susidaryti nekorektiška situacija. Pavyzdžiui, į sankryžą mašinos atvažiuoja penkiomis eismo juostomis, išvažiuoja tik viena. Arba iš pastato išeina daugiau žmonių nei įėjo. Matyt, kažkas lipo per langą....

Srautu grafe vadinsime visų iš pradinės viršūnės p išeinančių srautų sumą. 

Žemiau pateiktas pavyzdys iliustruoja grafą ir per jį praleidžiamus srautus �(111.1.a-b pav.). 

111.1.a-b pav.

Maksimalių srautų uždavinys apibrėžiamas taip: Duotas grafas, pasižymintis, aukščiau minėtomis savybėmis. Rasti tokius srautus s, kad srautas grafe būtų kuo didžiausias.

Srautų leidimo žemyn metodas

Metodas remiasi labai paprastu principu: vanduo teka iš aukštesnės vietos į žemesnę. Metodą aiškinsime vandens tekėjimo vamzdžiais terminais. Sakykime, kad vamzdžius jungia m mazgų. Kiekvienas mazgas turi baką, kuriame laikomas į jį atėjęs ir iš jo dar neišėjęs vanduo. Pradžioje visi mazgai, išskyrus pradinį yra nuliniame aukštyje. Pradinis mazgas pakeltas į aukštį m. Iš pradinio mazgo į su juo sujungtus mazgus išbėga tiek vandens, koks yra jungiančių vamzdžių pralaidumas. Tas vanduo subėga į bakus. 

Imamas mazgas, kurio bake yra vandens, jei galima, jis pakeliamas ir iš jo vanduo bėga į kitų vamzdžių bakus. Veiksmai kartojami, kol nei vieno mazgo (išskyrus pradinį ir galinį) bake nelieka vandens.

Algoritmą iliustruoja šie paveikslėliai (111.2.a-g pav.). Kursyvu rašomas viršūnės aukštis, pusjuodžiu šriftu – viršūnės bake esančių srautų kiekis. Pradinės ir galinės viršūnės į eilę neįtraukiame, nes baigus algoritmą pradinės bakas gali, o galinės privalo būti netuščias.

111.2.a-g pav.

Pateiksime formalų algoritmo užrašą Pseudo-Paskaliu..

procedure max_srautai (p, g, m;    { pradinė, galinė viršūnės, viršūnių skaičius }

                       sv;   { briaunų svoriai }

                       var s { srautai });

  var h,      { viršūnių aukščiai }

      bakas;  { bakas }

begin

  inicializuoti (h, bakas, s); { priskiriami nuliai }

  h[1] := m; { pradinės viršūnės aukštis }

  { paleidžiami srautai iš pradinės viršūnės }

  for u gretima viršūnei p do

    begin

      s[1, u] :=  sv[1, u];

      s[u, 1] := -sv[1, u];

      bakas[u] := sv[1, u];

    end;

  įtraukti į eilę visas viršūnes (išskyrus p ir g, , kurių bakai netušti }

  while eilė netuščia do

    begin

      imti viršūnę u iš eilės pradžios ir ją pašalinti iš eilės;

      while u bakas netuščias do

        begin

          pakelti (u, h);

          paleisti (u, s, bakas)

        end;

    end;

end; { max_srautai }



Apie darbą su eile galite pasiskaityti 83 uždavinio sprendimo aprašyme. Atskirai pateiksime procedūras pakelti ir paleisti. Šios procedūros atliekamos tik su tomis viršūnėmis, kurių bakai nėra tušti. 

Viršūnės u aukštis didinamas tik tada, jei visų viršūnių, į kurias iš jos galime paleisti srautus, aukščiai yra didesni arba lygūs u. Jei nors vienos viršūnės aukštis yra žemesnis, tuomet nėra reikalo didinti viršūnės aukštį: galima iš karto paleisti srautus.

Reiktų detalizuoti sąvoką galime paleisti srautus. Paleisti srautus iš u į v galima tada, jei sv[u, v] - s[u, v] > 0. Panagrinėkime keletą pavyzdžių (111.3.a-d pav.)

111.3.a-d pav.

a) atveju iš u į v galime paleisti tris srautus, kurio kiekis yra 3. b) atveju iš u į v srautų paleisti negalime. c) atveju galime paleisti keturis srautus. Iš u į v atėjo keturi srautai, jeigu neturime kur juos nukreipti, galime grąžinti atgal. d) atveju galime paleisti keturiolika srautų.

procedure pakelti (u; { keliama viršūnė }

                   var h { viršūnių aukščiai });

  var min;

begin

  inicializuoti min;

  for v yra grafo viršūnė do

    if (sv[u, v] - s[u, v] > 0) ( (h[u] <= h[v])

       then if h[v] < h[mini]

                 then mini := v

       else pakelti negalima;

  if galima pakelti

     then h[u] := h[mini] + 1

end; { pakelti }

Jei viršūnės u bakas netuščias, baką reikia ištuštinti. Randamos tokios viršūnės, kurios yra žemesnės nei u, ir į kurias galima nukreipti srautus ir tai padaroma. Iš principo, iš u srautus galima paleisti į bet kurią viršūnę, kuri yra žemiau nei u, tačiau panagrinėję aukščiau pateiktą pavyzdį, pamatysite, kad užtenka imti tik tas viršūnes, kurios yra lygiai vienetu žemesnės nei u.

procedure paleisti (u; var s; var bakas);

{ paleidžia viršūnės u bake esančius srautus }

  var kiekis;

begin

  for v yra grafo viršūnė do

    if (h[v] + 1 = h[u]) and (sv[u, v] - s[u, v] > 0)

       { jei v žemesnė ir į ją dar galima paleisti srautus }

       then begin

              kiekis := min (bakas[u], sv[u, v] - s[u, v]);

              bakas[u] := bakas[u] - kiekis;

              s[u, v] := s[u, v] + kiekis;

              s[v, u] := -s[u, v];

              bakas[v] := bakas[v] + kiekis;

              if v nėra pradinė ar galinė viršūnė ir jos bakas netuščias

                 then įtraukti į eilę v;

            end;

end; { paleisti }

Pateikiame Paskaliu užrašytą algoritmą.

const MAX_M = 100; { maksimalus miestų skaičius }

type lentelė1 = array [1..MAX_M] of integer;

     lentelė2 = array [1..MAX_M, 1..MAX_M] of integer;

     aibė ( set of 1..MAX_M



procedure max_srautai (m: integer; sv: lentelė2;

                       var s: lentelė2);

{ randa maksimalius srautus grafe }

  var e, { eilė }

      h, bakas: lentelė1;  { viršūnių aukščiai ir bakas }

      u, galva, uodega: integer;

      virš: aibė;



  procedure inicializuoti (var h, bakas: lentelė1;

                           var s: lentelė2 { srautai });

  { inicializuoja lenteles h, s, bakas }

    var u, v: integer;

  begin

    for u := 1 to MAX_M do

      begin

        h[u] := 0;

        bakas[u] := 0;

      end;

   for u := 1 to MAX_M do

     for v := 1 to MAX_M do

       begin

         s[u, v] := 0;

         s[v, u] := 0;

       end;

  end; { inicializuoti }



  procedure įtraukti_į_eilę (u: integer; var virš: aibė;

                             var e: lentelė1; var uodega: integer);

  { įtraukia į eilės pabaigą viršūnę u }

  begin

    virš :( virš + [u];

    e[uodega] := u;

    if uodega = MAX_M

       then uodega := 1

       else uodega := uodega + 1;

  end; { įtraukti_į_eilę }



  procedure pašalinti_iš_eilės (var galva: integer; 

                                var virš: aibė);

  begin

    virš :( virš - [u];

    if galva = MAX_M

       then galva := 1

       else galva := galva + 1

  end; { pašalinti_iš_eilės }



  procedure pakelti (u: integer; { keliama viršūnė }

                     var h: lentelė1 { viršūnių aukščiai });

  { padidinamas u aukštis }

  { vartojami globalūs kintamieji m, s, sv }

  var mini, v: integer;

      galima: boolean;

  begin

    galima := true; { ar galima pakelti viršūnę }

    mini := 0;

    for v := 1 to m do

      if (sv[u, v] - s[u, v] > 0)

         then if (h[u] <= h[v])

                 then begin

                        if mini = 0

                           then mini := v

                           else if h[v] < h[mini]

                                   then mini := v;

                      end

                 else galima := false;

    if galima

       then h[u] := h[mini] + 1

  end; { pakelti }



  function min (x, y: integer): integer;

  begin

    if x < y

       then min := x

       else min := y

  end; { min }



  procedure paleisti (u: integer;

                      var s: lentelė2; { srautai }

                      var bakas: lentelė1);

  { paleidžia viršūnės u bake esančius srautus }

  { vartojami globalūs kintamieji m, h, sv, e, uodega }

    var v, kiekis: integer;

  begin

    for v := 1 to m do

      if (h[v] + 1 = h[u]) and (sv[u, v] - s[u, v] > 0)

         { jei v žemesnė ir į ją dar galima paleisti srautus }

         then begin

                kiekis := min (bakas[u], sv[u, v] - s[u, v]);

                bakas[u] := bakas[u] - kiekis;

                s[u, v] := s[u, v] + kiekis;

                s[v, u] := -s[u, v];

                bakas[v] := bakas[v] + kiekis;

                if (bakas[v] > 0) and (v <> 1) 

                then if (v <> m) and (not (v in virš))

                     then įtraukti_į_eilę (v, virš, e, uodega);

              end;

    end; { paleisti }



begin

  inicializuoti (h, bakas, s);

  galva := 1; uodega := 1; { eilė tuščia }

  virš :( [];

  h[1] := m; { pradinės viršūnės aukštis }

  { paleidžiame srautus iš pradinės viršūnės }

  for u := 1 to m do

    if sv[1, u] > 0

       then begin

              s[1, u] :=  sv[1, u];

              s[u, 1] := -sv[1, u];

              bakas[u] := sv[1, u];

              if u <> m

                 then įtraukti_į_eilę (u, virš, e, uodega);

            end;

  while virš <> [] do

    begin

      u := e[galva];

      pašalinti_iš_eilės (galva, virš);

      while bakas[u] > 0 do

        begin

          pakelti (u, h);

          paleisti (u, s, bakas)

        end;

    end;

end; { max_srautai }

112. Kaip panešti milijoną. Uždavinys sprendžiamas dinaminio programavimo metodu.

Šiam, kaip ir daugeliui kitų uždavinių, sprendžiamų dinaminio programavimo metodu, kyla pagunda pritaikyti Godųjį algoritmą – išrikiuoti brangenybes jų verčių ir svorio santykio (t.y. rasti kiekvienos brangenybės svorio vieneto vertę) mažėjimo tvarka ir iš eilės imti vertingiausias brangenybes, kol dar galima panešti. Žemiau pateiktas pavyzdys parodo, kad toks algoritmas šio uždavinio nesprendžia:

Tarkime, kad vaikinas gali panešti maišą, sveriantį ne daugiau 50 svorio vienetų. Karalius turi 3 brangenybes, kurių svoriai ir vertės:



Nr.    Svoris              vertė               svorio vieneto vertė

1        10                   60                           6

2        20                 100                           5

3        30                 120                           4

Remiantis Godžiuoju algoritmu reikėtų imti 1 ir 2 brangenybes. Jų suminė vertė 160. Tačiau paėmę 2 ir 3 brangenybes, gausime didesnę suminę vertę – 220.

Uždavinį sprendžiame dinaminio programavimo metodu. Pirmasis žingsnis – jį suskaidyti į keletą mažesnių uždavinių. 

Pažymėkime V(i, sv) (, 1 ( i ( n, 1 ( sv ( s) maksimalią vertę siūlomų imti brangenybių, jei renkamasi tik iš brangenybių, kurių numeriai neviršija i, o visas svoris neviršija sv. Pavyzdžiui, V(40, 2) = 160 (žr. pavyzdį). Mat galime rinktis tik iš dviejų pirmųjų brangenybių, o išrinktų brangenybių svoris neturi viršyti 40. 

Taigi V (n, s) yra ieškoma brangenybių, kurias reikėtų imti, maksimali vertė, t.y. pradinis uždavinys. Su kiekviena brangenybe galimi du atvejai – ji dedama į maišą arba ne.

Pradėkime nagrinėti nuo n-osios brangenybės. Jeigu ją dedame į maišą, tai

V(n, s) ( V(n - 1, s - svn, n - 1) + ven;

čia svn – n-osios brangenybės svoris, o ven – jos vertė.

Jei tos brangenybės nededame į maišą, tai

V(n, s) ( V(n - 1, s).

Kaip reikėtų pasielgti su n-ąja brangenybe? Akivaizdu, kad reikia rinktis tą atvejį, kai gaunama didesnė vertė, t. y.

V(n, s) ( max (V(n - 1, s - svn) + ven, V(n - 1, s).

Pradinį uždavinį suskaidėme į du mažesnius. Norėdami rasti V(n, s), turime rasti V reikšmes su mažesniais parametrais. Turėdami šią rekursinę formulę, galėtume ir grįžimo metodu ieškoti sprendinio. Tačiau tai bus neefektyvu, nes bus ilgai perrenkama.

Taikykime Dinaminį metodą, kuris laiko sąnaudas pakeičia atminties sąnaudomis. Skaidant du gautus uždavinius į dar mažesnius, gali atsitikti taip, kad gausime bendrų dalių. Todėl vieną kartą apskaičiavę reiškinio V(i, sv) reikšmę ją įsimename. Jei tos pačios reikšmės prireiks antrą kartą – jos nebereikės skaičiuoti iš naujo. Tereikės pasiimti iš lentelės. 

Sąlygoje prašoma ne tik surasti imamų brangenybių bendrą vertę, bet ir pačių brangenybių numerius. Tam kiekvienam lentelės elementui V(i, sv) įvedame dar vieną atributą, kuris parodo imti, ar neimti brangenybės duotoje situacijoje. Atkreipiame dėmesį, kad esant vienokiam svoriui ta pati brangenybė imama, kitokiam – neįmanoma. 

Žemiau pateikta lentelė sudaryta ankstesniam pavyzdžiui.

V(sv, i)�1�2�3��0�0�0�0��1�0�0�0��10�60, imti�60, neimti�60, neimti��11�60, imti�60, neimti�60, neimti��…�…�…�…��20�60, imti�100, imti�100, neimti��…�…�…�…��30�60, imti�100, imti�160, neimti��…�…�…�…��50�60, imti�160, imti�220, imti��

Šią lentelę užpildėme visą (iš apačios į viršų), kad geriau iliustruotume pavyzdį. Algoritme lentelė pildoma rekursiškai iš viršaus į apačią, todėl užpildomi tik reikalingi lentelės elementai. Aukščiau esančioje lentelėje jie pažymėti pajuodintu šriftu.

const max_s = 100;   { maksimalus maišo svoris }

      max_b = 200;   { maksimalus brangenybių skaičius }

type lentelė = array [1..max_b, 1..max_s] of

                record

                  r: integer;  { maksimali vertė }

                  imti: boolean { ar imti brangenybę }

                end;

     duom = array [1..max_b] of integer;

var V: lentelė; { globalieji kintamieji }

    sv,  { brangenybių svoriai }

    ve: duom;  { brangenybių vertės }



procedure skaičiuoti(n, s: integer; var vertė: integer);

  var i, j: integer;

  function ieškoti (n, s: integer): integer;

  { rekursinė funkcija, pildanti lentelę V, iš karto randa maksimalią brangenybių, }

  { kurias galima panešti, vertę }

    var vert1, { maksimali vertė, jei neimame n-osios brangenybės }

        vert2: integer;{ maksimali vertė, jei tą brangenybę imame }

  begin

    if s = 0

    then begin   { jei svoris lygus nuliui }

           ieškoti := 0;

           exit

         end

       else if V[n, s].r <> -1  { arba jei ieškoma reikšmė apskaičiuota }

               then begin

                      ieškoti := V[n, s].r;

                      exit

                    end;

    { toliau nagrinėjame du atvejus: }

    { 1) jei n-osios brangenybės neimame }

    if n = 1

       then  vert1 := 0

       else  vert1 := ieškoti(n - 1, s);

    { 2) jei brangenybę imame }

    if (sv[n] <= s){ ją galime paimti, jei neviršysime svorio limito }

    then if n = 1

         then vert2 := ve[n]

         else vert2 := ve[n] + ieškoti(n-1, s-sv[n])

    else vert2 := -1;

    { pasirenkame vieną iš dviejų variantų }

    if vert1 >= vert2

       then begin

              ieškoti := vert1;

              V[n, s].r := vert1;

            end

       else begin

              ieškoti := vert2;

              V[n, s].r := vert2;

              V[n, s].imti := true

            end

  end; { ieškoti }

begin

  { užpildome lentelę pradinėmis reikšmėmis }

  for i := 1 to n do

    for j := 1 to s do

        begin

          V[i, j].imti := false;

          V[i, j].r := -1

        end;

  vertė := ieškoti (n, s)

end; { skaičiuoti }

113. Šachmatai. Uždavinys priklauso tokiai uždavinių kategorijai, kuriems padaryti nereikia žinoti jokio ypatingesnio algoritmo. Užtenka pasirinkti tinkamas duomenų struktūras ir tvarkingai padaryti. 

Perskaičius pradinius duomenis reikia įsiminti situaciją lentoje bei figūrų sąrašą. Apie kiekvieną lentos langelį žinome ar jame stovi figūra, ar ne. Jei stovi, įsiminsime figūros numerį. Kiekvieną figūrą apibūdina keli parametrai – jos spalva, tipas, langelis kuriame ji stovi. Žemiau pateikiame duomenų struktūras:



  { ----------------------------------- duomenų tipai ------------------------------- }

  type koordinatės = record

                       x, y: integer;

                     end;

       spalva = (juoda, balta);

       tipas = (karalius, valdovė, bokštas, rikis, 

                žirgas, pėstininkas);

       figūra = record

                  k: koordinates; { figūros koordinatės }

                  s: spalva;      { spalva }

                  t: tipas        { ir tipas }

                end;

       fig_sąr = array [0..64] of figūra;  { figūrų sąrašas }

                     { fig[0] – kuria figūra ir kur einama šacho atveju }

       lenta = array [1..8, 1..8] of integer;



{ ----------------------------------------- globalieji kintamieji ------------------------------------ }

  var n: integer;   { figūrų skaičius }

      fig: fig_sąr; { figūrų sąrašas }

      len: lenta;   { šachmatų lenta – kokiame langelyje kokia figūra }



Šį uždavinį moksleiviai sprendė olimpiadoje. Tačiau ne visi žino taisykles, nėra paprasta jas įsisavinti per trumpą laiką. Todėl sudėtingesniems atvejams (pavyzdžiui, kai pėstininkas yra pirmoje eilutėje, t. y. jau virtęs kita figūra ir pan.) nebuvo parinkti testai. Šių atvejų nenagrinėsime ir algoritme.

Pirma algoritmo dalis – tikrinimas, ar juodųjų karalius yra šachuojamas. Imamos visos baltos figūros ir paeiliui tikrinama, ar kuri nors figūra nepuola juodųjų karaliaus.

function ribose (x, y: integer): boolean;

{ ar koordinatės neišeina iš lentos ribų }

begin

  ribose := (x > 0) and (x < 9) and (y > 0) and (y < 9)

end; { ribose }



{ ----------------------------------- ar karalius šachuojamas --------------------------------- }

function yra_fig (x, y: integer; tp: tipas; sp: spalva)

                                               : boolean;

{ ar pozicijoje (x, y) yra „tp“ tipo ir „sp“ spalvos figūra }

begin

  if len[x, y] = 0

  then yra_fig := false

  else yra_fig := (fig[len[x,y]].t = tp) and 

                  (fig[len [x,y]].s = sp)

end; { yra_fig }



function šach_krypt (kor: koordinatės; dx, dy: integer)

                                               : boolean;

{ tikrina, ar figūra, galinti judėti nurodyta kryptimi šachuoja karalių; }

{ jos judėjimo kryptį nusako dx ir dy – koordinačių pokyčiai }

begin

  kor.x := kor.x + dx;

  kor.y := kor.y + dy;

  while ribose(kor.x,kor.y) and (len[kor.x,kor.y] = 0) do

    begin                   { einame, kol sutinkame figūrą ar }

      kor.x := kor.x + dx;  { prieiname lentos kraštą }

      kor.y := kor.y + dy

    end;

  if ribose (kor.x, kor.y)

     then if yra_fig (kor.x, kor.y, karalius, juoda)

             then šach_krypt := true

             else šach_krypt := false { sutikome ne karalių }

     else šach_krypt := false { išėjome iš ribų nesutikę figūros }

end;  { šach_krypt }



function šach_pėst (kor: koordinates): boolean;

begin

  šach_pėst := ribose(kor.x-1, kor.y+1) and

            yra_fig(kor.x-1, kor.y+1, karalius, juoda) or

            ribose(kor.x+1, kor.y+1) and

            yra_fig(kor.x+1, kor.y+1, karalius, juoda)

end; { šach_pėst }



function šach_rikis (kor: koordinates): boolean;

begin

  šach_rikis := 

    šach_krypt(kor,-1,-1) or šach_krypt(kor,1,-1) or

    šach_krypt(kor,-1,1)  or šach_krypt(kor,1,1)

end; { šach_rikis }



function šach_bokštas (kor: koordinates): boolean;

begin

  šach_bokštas := 

   šach_krypt(kor,0,-1) or šach_krypt(kor,0,1) or

   šach_krypt(kor,-1,0) or šach_krypt(kor,1,0)

end; { šach_bokštas }



function šach_valdovė (kor : koordinates): boolean;

begin

  šach_valdovė := šach_rikis(kor) or šach_bokštas(kor)

end;  { šach_valdovė }



function šach_žirgas (kor: koordinates): boolean;

  var i, j: integer;

      šach: boolean;

begin

  šach := false;

  for i:=-2 to 2 do

   for j:=-2 to 2 do

    if (abs(i)+abs(j)=3) and ribose(kor.x+i, kor.y+j) and

       yra_fig(kor.x+i, kor.y+j, karalius, juoda)

       then šach := true;

  šach_žirgas := šach

end; { šach_žirgas }



function šachas: boolean;

{ tikrina, ar juodųjų karalius šachuojamas }

{ n, fig, len – globalieji kintamieji }

  var i: integer;

      šach: boolean;

begin

  šach := false;

  i := 1;

  while (i <= n) and not šach do

    begin

      with fig[i] do    { tikrina, ar nešachuoja kuri nors figūra }

        if s = balta

           then case t of

                  pėstininkas: šach := šach_pėst(k);

                  žirgas     : šach := šach_žirgas(k);

                  rikis      : šach := šach_rikis(k);

                  bokštas    : šach := šach_bokštas(k);

                  valdovė    : šach := šach_valdovė(k)

                end;

      i := i + 1

    end;

  šachas := šach

end; { šachas }



Jei juodųjų karalius šachuojamas, reikia nustatyti, ar juodiesiems matas, ar mato dar galima išvengti. Paeiliui imamos visos juodos figūros ir su kiekviena figūra atliekami visi galimi ėjimai bei kirtimai. Atlikus ėjimą, tikrinama, ar karalius tebėra šachuojamas, jei taip – figūra gražinama į ankstesnę padėtį, atstatoma nukirsta figūra (jei buvo kertama) ir bandoma daryti kitą ėjimą su ta pačia figūra, kol nebelieka galimų ėjimų. Tada bandoma eiti kita juoda figūra, kol išbandomi visi variantai.

{ --------------------------------------- tikrinama, ar matas --------------------------------- }

function yra (x, y: integer; sp: spalva): boolean;

{ ar pozicijoje (x, y) yra bet kokia „sp“ spalvos figūra }

begin

  if len[x, y] = 0

     then yra := false

     else yra := fig[len [x, y]].s = sp

end; { yra }



function pasiseka (f, x, y: integer): boolean;

{ ar pasiseks išvengti mato, paėjus figūra f į langelį (x, y) }

  var buv_f: integer;

      buv_k: koordinates;

      nukirsta: figūra;

begin

  { patikriname, ar nebandoma kirsti savo spalvos figūros }

  buv_f := len[x, y]; { sužinome, kokia figūra stovi kertamame }

  if (buv_f = 0) or (fig[buv_f].s = balta)  { langelyje }

  then with fig[f] do

         begin

           buv_k := k;         { įsimename buvusias koordinates }

           len[k.x, k.y] := 0; { atliekame ėjimą }

           k.x := x; k.y := y;

           len[x, y] := f;

           if buv_f <> 0       { bei nukertame figūrą }

              then begin

                     nukirsta := fig[buv_f];

                     fig[buv_f] := fig[n];

                     n := n - 1

                   end;

           if not šachas

              then begin { šacho išvengta }

                     pasiseka := true;

                     fig[0] := fig[f];

                   end

              else begin { grąžiname atgal figūras }

                     k.x := buv_k.x; k.y := buv_k.y; 

                     len[k.x, k.y] := f;

                     len[x,y] := buv_f; {atstatome nukirstą figūrą}

                     if buv_f <> 0

                        then begin

                               n := n + 1;

                               fig[n] := fig[buv_f];

                               fig[buv_f] := nukirsta

                             end;

                     pasiseka := false

                   end

         end

  else pasiseka := false

end; { pasiseka }



function eik_krypt (f, dx, dy: integer): boolean;

{ figūra f, galinti juda nurodyta kryptimi, kol apgina karalių arba nebegali toliau eiti }

{ jos judėjimo kryptį nusako dx ir dy – koordinačių pokyčiai }

  var saugu: boolean;

      kor: koordinatės;

begin

  saugu := false;

  kor.x := fig[f].k.x + dx;

  kor.y := fig[f].k.y + dy;

  while ribose(kor.x,kor.y) and (len[kor.x,kor.y]=0) and 

        not saugu do

    begin                        { einame, kol sutinkame figūrą ar }

      saugu := pasiseka(f,kor.x,kor.y); { prieiname lentos kraštą }

      kor.x := kor.x + dx;

      kor.y := kor.y + dy

    end;

  if ribose (kor.x, kor.y)  { jei sutikome figūrą }

     then saugu := pasiseka (f, kor.x, kor.y);

  eik_krypt := saugu

end; { eik_krypt }



function ėj_rikis (f: integer): boolean;

begin

  ėj_rikis := eik_krypt(f,-1,-1) or eik_krypt(f,1,-1) or

              eik_krypt(f,-1,1)  or eik_krypt(f,1,1)

end; { ėj_rikis }



function ėj_bokštas (f: integer): boolean;

begin

  ėj_bokštas := eik_krypt(f,0,-1) or eik_krypt(f,0,1) or

                eik_krypt(f,-1,0) or eik_krypt(f,1,0)

end; { ėj_bokštas }



function ėj_valdovė (f: integer): boolean;

begin

  ėj_valdovė := ėj_rikis(f) or ėj_bokštas(f)

end; { ėj_valdovė }



function ėj_žirgas (f: integer): boolean;

  var i, j: integer;

      kor: koordinatės;

      saugu: boolean;

begin

  kor := fig[f].k; { figūros koordinatės }

  saugu := false;

  for i:=-2 to 2 do

    for j:=-2 to 2 do

      if not saugu and (abs(i)+abs(j) = 3) and 

        ribose(kor.x+i,kor.y+j) and 

        pasiseka f,kor.x+i,kor.y+j)

        then saugu := true;

  ėj_žirgas := saugu

end; { ėj_žirgas }



function ėj_karalius (f: integer): boolean;

  var i, j: integer;

      kor: koordinatės;

      saugu: boolean;

begin

  kor := fig[f].k;

  saugu := false;

  for i := -1 to 1 do

    for j := -1 to 1 do

      if not saugu and (abs(i) + abs(j) <> 0) and

         ribose(kor.x+i, kor.y+j) and 

         pasiseka(f, kor.x+i, kor.y+j)

         then saugu := true;

  ėj_karalius := saugu

end; { ėj_karalius }



function ėj_pėst (f: integer): boolean;

  var kor: koordinates;

begin

  kor := fig[f].k;

  ėj_pėst :=

   ribose(kor.x-1,kor.y-1) and yra(kor.x-1,kor.y-1,balta)

   and pasiseka(f,kor.x-1,kor.y-1)    { kertama į dešinę }

   or

   ribose(kor.x+1,kor.y-1) and yra(kor.x+1,kor.y-1,balta) 

   and pasiseka(f,kor.x+1,kor.y-1)    { kertama į kairę }

   or

   ribose(kor.x,kor.y-1) and (len[kor.x,kor.y-1] = 0)

   and pasiseka(f,kor.x,kor.y-1) { einama pirmyn per vieną langelį }

   or

   (kor.y=7) and (len[kor.x,5]=0) and (len[kor.x,6]=0)

   and pasiseka(f,kor.x,5) { einama pirmyn per du langelius }

end; { ėj_pėst }



function matas: boolean;

{ tikrina, ar juodųjų karaliui matas }

{ n, len, fig – globalieji kintamieji }

  var i: integer;

      saugu: boolean;

begin

  saugu := false;

  i := 1;

  while (i <= n) and not saugu do

    begin  { tikrinsime, ar galime išsigelbėti paėję i-tąją figūra }

      with fig[i] do

        if s = juoda  { jei tai juodoji figūra, bandome ja paeiti }

        then case t of  { sėkmingas ėjimas įsimenamas fig[0] }

               pėstininkas: saugu := ėj_pėst(i);

               žirgas     : saugu := ėj_žirgas(i);

               rikis      : saugu := ėj_rikis(i);

               bokštas    : saugu := ėj_bokštas(i);

               valdovė    : saugu := ėj_valdovė(i);

               karalius   : saugu := ėj_karalius(i)

             end;

        i := i + 1;

    end;

  matas := not saugu

end; { matas }



Nepateikėme tik pradinių duomenų įvedimo (įvedamos n, fig ir len reikšmės) bei pagrindinės programos dalies. Žemiau pateikiame pradinių duomenų rinkinius. Taupant vietą figūrų padėtys lentelėje surašytos į vieną eilutę ir atskirtos kableliu. 



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�7

bK14, bB15, bR51, bB71

jB18, jK85, jB86�jB15�Lentoje tik trijų rūšių figūros – karaliai, bokštai, rikis��2�10

bK15, bP24, bR31, bV35,

bR38, bZ57, jP63, jP72

jB84, jK85�MATAS�Lentoje yra visų tipų figūrų��3�11

bK15, bB17, bP25, bV46

bZ54, jZ58, jR62, jP74

jV83, jB85, jK86�jZ46 �Kitas galimas atsakymas yra jZ66;

Lentoje yra visų tipų figūrų ��4�8

bK35, bP44, bV45, jP73

jP74, jB84, jB85, jK86�NEŠACHUOJAMA�Lentoje nėra žirgų ir rikių��5�10

bK41, bP42, bB51, bR83

bV84, jZ37, jB38, jK48

jP77, jR78�jP75�Lentoje yra visų tipų figūrų��

